

ЗАНЯТИЕ 16

Бухгалтерский учет

ПРОДОЛЖИТЕЛЬНОСТЬ

Ориентировочная продолжительность занятия – 1 час 50 минут.

План видов характеристик в бухгалтерском учете	524
Добавление плана видов характеристик	525
Что такое «План счетов»	528
Добавление плана счетов	529
Что такое регистр бухгалтерии	534
Добавление регистра бухгалтерии	535
Доработка приходной накладной	537
Доработка документа «Оказание услуги»	542
Оборотно-сальдовая ведомость	545
В режиме «Конфигуратор»	545
Запрос для набора данных	545
Ресурсы	547
Параметры	547
Настройки	548
В режиме «1С:Предприятие»	550
Контрольные вопросы	551

На этом занятии мы проиллюстрируем возможность ведения бухгалтерского учета средствами «1С:Предприятия». В рамках этого занятия мы не будем объяснять и рассматривать основы бухгалтерского учета. Поэтому если у вас нет знаний бухгалтерии, то, конечно, лучше сначала прочитать какую-нибудь популярную литературу о том, как вообще устроен бухгалтерский учет в нашей стране.

Для организации бухгалтерского учета мы используем уже знакомый нам объект конфигурации – план видов характеристик и два новых объекта – *План счетов* и *Регистр бухгалтерии*.

Регистр бухгалтерии будет использоваться нами для накопления данных о совершенных хозяйственных операциях.

С помощью плана счетов мы будем описывать счета, в разрезе которых ведется учет, а план видов характеристик будет служить для описания объектов аналитического учета, в разрезе которых должен вестись учет на счетах.

Сразу оговоримся, что план счетов, который мы будем использовать в нашей учебной конфигурации, очень сильно упрощен. Он содержит всего несколько условных счетов, которые, однако, позволят нам познакомиться с основными методами организации бухгалтерского учета средствами «1С:Предприятия».

План видов характеристик в бухгалтерском учете

Объект конфигурации План видов характеристик был подробно рассмотрен нами на предыдущем занятии (см. «Что такое план видов характеристик» на стр. 473), поэтому сейчас мы проиллюстрируем только использование этого объекта в контексте бухгалтерского учета.

Бухгалтерский учет, как правило, подразумевает ведение аналитического учета на большинстве счетов. Для обозначения разрезов аналитического учета мы будем использовать термин *виды субконто*. То есть на каждом счете учет может вестись в разрезе нескольких видов субконто.

А для обозначения конкретных объектов аналитического учета мы будем использовать термин *субконто*.

Например, на 41 счете (Товары) учет ведется обычно в разрезе Номенклатуры и Складов, которые являются видами субконто.

А вот конкретная номенклатура Паста шоколадная и конкретный склад Основной, указанные для некоторой проводки по 41 счету, – это субконто.

Так вот, частным случаем использования плана видов характеристик является применение его для описания видов субконто. То есть все разрезы аналитического учета описываются в соответствующем плане видов характеристик, и там же задаются типы значений, которые могут принимать те или иные субконто.

Добавление плана видов характеристик

В режиме «Конфигуратор»

Приступим к созданию плана видов характеристик, который будет содержать описания разрезов аналитического учета – видов субконто.

Откроем конфигуратор и добавим новый объект конфигурации План видов характеристик. Зададим его имя – ВидыСубконто. На закладке Подсистемы укажем, что план счетов будет отображаться в подсистеме Бухгалтерия.

Поскольку нам понадобится некий вспомогательный справочник, в котором пользователи будут осуществлять «свободное творчество» по созданию значений новых объектов аналитического учета, добавим объект конфигурации Справочник и назовем его Субконто.

Затем на закладке Владельцы укажем, что этот справочник будет подчинен плану видов характеристик ВидыСубконто.

Для этого на закладке Владельцы нажмем кнопку Редактировать элемент списка и выберем в качестве владельца справочника план видов характеристик ВидыСубконто (рис. 16.1).

Закроем окно редактирования справочника и вернемся к нашему плану видов характеристик.

На закладке Основные установим свойство Тип значения характеристик. Нажмем кнопку выбора и зададим составной тип данных следующим образом (рис. 16.2):

- СправочникСсылка.Клиенты,
- СправочникСсылка.Номенклатура,
- СправочникСсылка.Субконто.




Рис. 16.1. Окно редактирования справочника «Субконто»




Рис. 16.2. Выбор типа значения характеристик плана вида характеристик

Бухгалтерия нашего ООО «На все руки мастер» ведет учет движения денежных средств только в разрезе материалов и клиентов, но не исключено, что в дальнейшем понадобится дополнительная аналитика (поэтому мы и используем справочник Субконто).

Обратите внимание, что тот справочник, который будет использован в качестве дополнительных значений характеристик, тоже должен входить в составной тип данных типа значений характеристик, иначе конфигуратор выдаст сообщение об ошибке.

Затем укажем, что дополнительные значения характеристик будут находиться в справочнике Субконто.

После этого перейдем на закладку Прочее и, нажав кнопку Предопределенные, начнем ввод предопределенных значений плана видов характеристик (рис. 16.3). То есть тех видов аналитического учета, в разрезе которых мы будем вести учет.




Рис. 16.3. Предопределенные виды характеристик

Нажимая кнопку Добавить, создадим предопределенный вид субконто Материалы с кодом 000000001 и типом СправочникСсылка.Номенклатура.

Затем создадим вид субконто Клиенты с кодом 000000002 и типом СправочникСсылка.Клиенты.

На этом создание видов субконто завершено, и мы можем перейти к знакомству со следующим объектом конфигурации, который будет использован нами, – План счетов.

Что такое «План счетов»

Объект конфигурации *План счетов* предназначен для описания структуры хранения информации о совокупности синтетических счетов предприятия, которые созданы для группировки данных о его хозяйственной деятельности.

На основе объекта конфигурации *План счетов* платформа создает в базе данных таблицы, в которых будет храниться информация о том, какие счета и каким образом будет использовать предприятие.

Это может быть система бухгалтерских счетов, установленная государством, план управленческих счетов или произвольный набор счетов, используемых для анализа тех или иных видов деятельности предприятия.

План счетов в системе «1С:Предприятие» поддерживает иерархию субсчетов: к каждому счету первого уровня может быть открыто несколько субсчетов, которые, в свою очередь, могут иметь свои субсчета, и так далее.

Например, законодательно утвержденный план счетов для ведения бухучета в России имеет следующий вид (рис. 16.4).

Код	Наименование	Заб.	Акт.	Вал.	Кол.	Субконтно 1	Субконтно 2	Субконтно 3
01	Основные средства		А			Основные средс...		
01.01	Основные средства в организа...		А			Основные средс...		
01.09	Выбытие основных средств		А			Основные средс...		
02	Амортизация основных средств		П			Основные средс...		
02.01	Амортизация основных средств...		П			Основные средс...		
02.02	Амортизация основных средств...		П			Основные средс...		
03	Доходные вложения в материал...		А			Контрагенты	Основные средс...	
03.01	Материальные ценности в орга...		А			Основные средс...		
03.02	Материальные ценности предос...		А			Контрагенты	Основные средс...	
03.03	Материальные ценности предос...		А			Контрагенты	Основные средс...	
03.04	Прочие доходные вложения		А			Контрагенты	Основные средс...	
03.09	Выбытие материальных ценнос...		А			Основные средс...		
04	Нематериальные активы		А			Нематериальны...		
04.01	Нематериальные активы орган...		А			Нематериальны...		
04.02	Расходы на научно-исследовате...		А			Нематериальны...		
05	Амортизация нематериальных а...		П			Нематериальны...		
07	Оборудование к установке		А		✓	Номенклатура	Оклады	Партии
08	Вложения во внеоборотные акт...		А			Объекты строит...	(06) Статьи затрат	
08.01	Приобретение земельных участ...		А			Объекты строит...	(06) Статьи затрат	
08.02	Приобретение объектов природ...		А			Объекты строит...	(06) Статьи затрат	
08.03	Строительство объектов основ...		А			Объекты строит...	(06) Статьи затрат	Способы строит...

Рис. 16.4. Российский план счетов

По любому счету или субсчету может вестись аналитический учет в разрезе субконто, описанных в плане видов характеристик. Связь между планом счетов и планом видов характеристик задается разработчиком на этапе конфигурирования.

Для описания используемых субконто система создает в плане счетов специальную табличную часть **ВидыСубконто**, которая не видна в конфигураторе (но доступна средствами встроенного языка).

Для каждого счета есть возможность задать несколько видов учета (например, количественный и валютный). Виды учета задаются при помощи подчиненных объектов конфигурации *признак учета*.

Также существует возможность определить несколько видов учета субконто (например, суммовой, валютный или количественный). Виды учета субконто задаются при помощи подчиненных объектов конфигурации *признак учета субконто*.

УЗНАЙ БОЛЬШЕ!

О структуре объектов встроенного языка, предназначенных для работы с планами счетов, можно прочитать в разделе «Краткий справочник разработчика. Планы счетов» на стр. 920.

Помимо всего вышеперечисленного каждый счет может иметь набор свойств, которые задаются в качестве реквизитов объекта конфигурации **План счетов**. Они позволяют определять уникальные свойства элементов плана счетов (например, реквизит **ЗапретитьИспользоватьВПроводках**).

Добавление плана счетов

В режиме «Конфигуратор»

Приступим к созданию плана счетов ООО «На все руки мастер».

Как мы говорили в начале этого занятия, бухгалтерский учет в нашем ООО «На все руки мастер» сильно упрощен. Поэтому план счетов, по которому работает бухгалтерия, содержит всего четыре счета:

- Товары,
- РасчетыСПоставщиками,
- Капитал,
- Дебиторская задолженность.

Добавим новый объект конфигурации План счетов. Присвоим ему имя – Основной. Свойство Представление списка зададим как Основной план счетов. На закладке Подсистемы укажем, что план счетов будет отображаться в подсистеме Бухгалтерия. На закладке Данные выделим группу реквизитов Признаки учета и, нажав кнопку Добавить, создадим признак учета Количественный (рис. 16.5).




Рис. 16.5. Создание реквизита плана счетов в группе «Признаки учета»

Перейдем на закладку Субконто и укажем, что виды субконто для этого плана счетов будут находиться в плане видов характеристик ВидыСубконто.

Максимальное количество субконто на счете установим равным двум.

Также создадим признак учета субконто Количественный (рис. 16.6).




Рис. 16.6. Установка свойств «Субконто для плана счетов»

Затем откроем закладку Прочее. Нажмем кнопку **Предопределенные** и создадим четыре предопределенных счета (при создании каждого счета, перед тем как нажать кнопку **Добавить**, нужно выделить корень структуры счетов – строку **Счета**):

- Товары, код 41, активный, с количественным учетом в разрезе материалов (рис. 16.7).





Рис. 16.7. Предопределенный счет «Товары»

- РасчетыСПоставщиками, код 60, активный/пассивный (рис. 16.8).



The dialog box 'Предопределенный счет' (Predefined Account) is shown. It contains the following fields and options:

- Родитель: Счета
- Имя: РасчетыСПоставщиками
- Код: 60
- Наименование: Расчеты с поставщиками
- Вид: Активный/Пассивный (dropdown menu)
- ☐ Забалансовый
- Порядок: (empty field)

Below these fields is a table for 'Признак: учета' (Accounting Significance):

Признак: учета	Учитывать
Количественный	☐


At the bottom, there is a section for 'Вид субконто' (Subaccount Type) with a table:

Вид субконто	Только обороты	Количественный
Клиенты	☐	☐

Buttons at the bottom: OK, Отмена, Справка.

Рис. 16.8. Предопределенный счет «РасчетыСПоставщиками»

- ДебиторскаяЗадолженность, код 62, активный/пассивный, в разрезе клиентов (рис. 16.9).



The dialog box 'Предопределенный счет' (Predefined Account) is shown. It contains the following fields and options:

- Родитель: Счета
- Имя: ДебиторскаяЗадолженность
- Код: 62
- Наименование: Дебиторская задолженность
- Вид: Активный/Пассивный (dropdown menu)
- ☐ Забалансовый
- Порядок: (empty field)

Below these fields is a table for 'Признак: учета' (Accounting Significance):

Признак: учета	Учитывать
Количественный	☐

At the bottom, there is a section for 'Вид субконто' (Subaccount Type) with a table:

Вид субконто	Только обороты	Количественный
Клиенты	☐	☐

Buttons at the bottom: OK, Отмена, Справка.

Рис. 16.9. Предопределенный счет «ДебиторскаяЗадолженность»

- Капитал, код 90, активный/пассивный (рис. 16.10).

Рис. 16.10. Предопределенный счет «Капитал»

В результате план счетов нашего ООО «На все руки мастер» будет выглядеть следующим образом (рис. 16.11).

Имя	Код	Наименование	Вид	Заб...	Пор...	Количес...	Субконто 1	С...
Счета								
Товары	41	Товары	Активный			✓	Материалы	
Расчеты с поставщиками	60	Расчеты с поставщиками	Активный/Пассивный					
Дебиторская задолженность	62	Дебиторская задолженность	Активный/Пассивный				Клиенты	
Капитал	90	Капитал	Активный/Пассивный					

Рис. 16.11. План счетов «Основной»

Теперь мы можем перейти к знакомству с последним объектом конфигурации, который понадобится нам для организации бухгалтерского учета, – *Регистром бухгалтерии*.

УЗНАЙ БОЛЬШЕ!

Для плана счетов можно установить свойство «Автопорядок по коду». Это свойство используется для того, чтобы указать системе, что упорядочивание по полю «Порядок» должно всегда подставляться в тех случаях, когда пользователь или разработчик выбирает упорядочивание по коду. Его нужно использовать прежде всего тогда,

когда с точки зрения пользователя нужно упорядочивать план счетов по коду с учетом разделителей кода счета. Например, если счета 10.11 и 10.2 упорядочивать по коду счета, то счета будут располагаться так:

10.11

10.2

Это правильно с точки зрения сортировки строк, но не соответствует логическому смыслу кодов.

Но если заданы значения поля «Порядок» 10.11 и 10. 2 (перед 2 – пробел) и установлено свойство «Автопорядок по коду», то при выборе упорядочивания по коду пользователь будет фактически получать порядок, учитывающий разделители:

10.2

10.11

Если свойство не устанавливать, то нужно будет в явном виде выбирать упорядочивание по полю «Порядок».

Что такое регистр бухгалтерии

Объект конфигурации *Регистр бухгалтерии* предназначен для описания структуры накопления данных, учет которых ведется исходя из некоторого плана счетов. На основе объекта конфигурации *Регистр бухгалтерии* платформа создает в базе данных таблицу, в которой будут накапливаться данные о хозяйственных операциях, отображаемых в бухгалтерском учете.

По своему виду регистр бухгалтерии напоминает регистр накопления – он также имеет ресурсы, может иметь измерения и реквизиты.

Измерения позволяют разделять ведение учета (например, используя измерение *Организация*, можно вести учет в разрезе нескольких юридических лиц).

Реквизиты служат признаком, по которому одни записи регистра можно отделить от других (например, в качестве реквизита может использоваться номер журнала, что позволит отбирать проводки, имеющие одинаковый смысл).

Значительное отличие от регистра накопления заключается в том, что регистр бухгалтерии имеет жесткую связь с используемым планом

счетов. Поэтому каждая запись регистра бухгалтерии содержит дополнительные поля, определяемые настройкой используемого плана счетов.

Например, запись регистра может содержать дополнительные поля для указания корреспондирующих счетов, сумм, объектов аналитического учета (субконто), количества, вида валюты и т. д.

Кроме этого, отличительной чертой регистра бухгалтерии является возможность поддержки механизма двойной записи, при которой каждая запись регистра содержит обязательные поля для указания счета дебета и счета кредита.

УЗНАЙ БОЛЬШЕ!

О структуре объектов встроенного языка, предназначенных для работы с регистрами бухгалтерии, можно прочитать в разделе «Краткий справочник разработчика. Регистры бухгалтерии» на стр. 935.

Добавление регистра бухгалтерии

В режиме «Конфигуратор»

Создадим новый объект конфигурации Регистр бухгалтерии. Зададим его имя – Управленческий. Свойство Расширенное представление списка зададим как Движения в регистре Управленческий. Укажем, что с ним будет связан план счетов Основной. Установим флажок Корреспонденция (рис. 16.12).




Рис. 16.12. Основные свойства регистра бухгалтерии

Флажок Корреспонденция будет говорить о том, что создаваемый нами регистр поддерживает корреспонденцию. Это означает, что каждая запись регистра имеет дебетовую и кредитовую часть, что позволит нам получать информацию не только об остатках и оборотах по счетам, но и о корреспонденциях между счетами.

Регистры, не поддерживающие корреспонденцию, применяются тогда, когда не нужно использовать принцип двойной записи, регламентированный в бухгалтерском учете, и, соответственно, контролировать баланс хозяйственных средств и их источников.

На закладке Подсистемы укажем, что регистр будет отображаться в подсистеме Бухгалтерия.

Теперь перейдем на закладку Данные и создадим два ресурса:

- Сумма, длина 15, точность 2, Балансовый;
- Количество, длина 15, точность 3, Небалансовый, признак учета – Количественный, признак учета субконто – Количественный (рис. 16.13).

Свойства: Количество

Имя: Количество

Синоним: Количество

Комментарий:

Тип: Число

Длина: 15

Точность: 3

Неотрицательное: ☐

Использование: Полнотекстовый поиск: Использовать

Представление: Формат: Число

Формат редактирования: Число

Подсказка: Количество

Выделять отрицательные: ☐

Минимальное значение:

Максимальное значение:

Проверка заполнения: Не проверять

Быстрый выбор: Авто

Создание при вводе: Авто

Связь по типу: Балансовый: Не

Признак учета: Количественный

Признак учета субконто: Количественный

Сохранять баланс: ☒

Рис. 16.13. Свойства ресурса «Количество» регистра бухгалтерии

В заключение сделаем видимой команду для открытия списка регистра в разделе Бухгалтерия и расположим ее после всех созданных нами ранее регистров.

Откроем командный интерфейс подсистемы Бухгалтерия из контекстного меню этой подсистемы, выделив ее в списке подсистем в дереве объектов конфигурации. В группе Панель навигации.Обычное включим видимость у команды Управленческий и мышью перетащим ее в группу Панель навигации.См. также.

На этом создание нашего регистра бухгалтерии завершено.

Теперь настало время познакомиться с тем, каким образом используется созданный нами регистр бухгалтерии Управленческий.

Доработка приходной накладной

Что нас ждет дальше?

Сначала мы доработаем оба наши документа (ПриходнаяНакладная и ОказаниеУслуги) так, чтобы они «поставляли» данные не только для регистров накопления, но и для регистра бухгалтерии.

Затем мы создадим бухгалтерский отчет Оборотно-сальдовая ведомость, который будет показывать нам состояние товародвижения в ООО «На все руки мастер», основываясь на данных регистра бухгалтерии.

При проведении наши документы будут создавать следующие бухгалтерские проводки – записи в регистре бухгалтерии (таблица 16.1).

Таблица 16.1. Проводки, создаваемые документами

Документы	Проводки				
	Дебет		Кредит		Сумма
Приходная накладная	41	Товары	60	Расчеты с поставщиками	Стоимость
Оказание услуги	62	Дебиторская задолженность	90	Капитал	Выручка
	90	Капитал	41	Товары	Стоимость

Эти проводки мы отразим при создании движений документов в регистре бухгалтерии.

Итак, сначала изменим процедуру проведения документа ПриходнаяНакладная, а затем в режиме 1С:Предприятие перепроведем все эти документы, чтобы отработал новый, измененный нами алгоритм проведения документа ПриходнаяНакладная.

В режиме «Конфигуратор»

В окне редактирования объекта конфигурации Документ ПриходнаяНакладная перейдем на закладку Движения. В списке регистров отметим, что документ будет создавать теперь движения и по регистру бухгалтерии Управленческий. Перейдем на закладку Прочее и откроем модуль объекта. Откроем процедуру обработчика события ОбработкаПроведения. В самом конце цикла перед строкой КонецЦикла добавим строки кода, создающие движения регистра Управленческий (листинг 16.1).

Листинг 16.1. Движения документа «ПриходнаяНакладная» (фрагмент)

```
// Регистр Управленческий
Движение = Движения.Управленческий.Добавить();
Движение.СчетДт = ПланыСчетов.Основной.Товары;
Движение.СчетКт = ПланыСчетов.Основной.РасчетыСПоставщиками;
Движение.Период = Дата;
Движение.Сумма = ТекСтрокаМатериалы.Сумма;
Движение.КоличествоДт = ТекСтрокаМатериалы.Количество;
Движение.СубконтоДт[ПланыВидовХарактеристик.ВидыСубконто.Материалы] =
ТекСтрокаМатериалы.Материал;
```

Перед началом цикла установим свойство Записывать набора записей регистра Управленческий в значение Истина для записи изменений регистра в базу данных.

В результате процедура ОбработкаПроведения будет выглядеть следующим образом (листинг 16.2).

Листинг 16.2. Движения документа «ПриходнаяНакладная»

Процедура ОбработкаПроведения(Отказ, Режим)

```
Движения.ОстаткиМатериалов.Записывать = Истина;
Движения.СтоимостьМатериалов.Записывать = Истина;
Движения.Управленческий.Записывать = Истина;
```

Для Каждого ТекСтрокаМатериалы Из Материалы Цикл

```
// Регистр ОстаткиМатериалов Приход
Движение = Движения.ОстаткиМатериалов.Добавить();
```



```
Движение.ВидДвижения = ВидДвиженияНакопления.Приход;  
Движение.Период = Дата;  
Движение.Материал = ТекСтрокаМатериалы.Материал;  
Движение.НаборСвойств = ТекСтрокаМатериалы.НаборСвойств;  
Движение.Склад = Склад;  
Движение.Количество = ТекСтрокаМатериалы.Количество;  
  
// Регистр Стоимость Материалов Приход  
Движение = Движения.СтоимостьМатериалов.Добавить();  
Движение.ВидДвижения = ВидДвиженияНакопления.Приход;  
Движение.Период = Дата;  
Движение.Материал = ТекСтрокаМатериалы.Материал;  
Движение.Стоимость = ТекСтрокаМатериалы.Сумма;  
  
// Регистр Управленческий  
Движение = Движения.Управленческий.Добавить();  
Движение.СчетДт = ПланыСчетов.Основной.Товары;  
Движение.СчетКт = ПланыСчетов.Основной.РасчетыСПоставщиками;  
Движение.Период = Дата;  
Движение.Сумма = ТекСтрокаМатериалы.Сумма;  
Движение.КоличествоДт = ТекСтрокаМатериалы.Количество;  
Движение.СубконтоДт[ПланыВидовХарактеристик.ВидыСубконто.Материалы] =  
ТекСтрокаМатериалы.Материал;  
  
КонецЦикла;  
  
КонецПроцедуры
```

Прокомментируем этот код.

Как видите, движения для регистра бухгалтерии формируются таким же образом, как и для регистра накопления.

Но платформа (при создании таблицы хранения данных) добавила к созданным нами реквизитам регистра еще ряд полей, которые явились следствием использования плана счетов Основной.

Прежде всего, это поля СчетДт и СчетКт. В этих полях указываются счета, дебет и кредит которых затрагивает данная проводка.

Кроме этого, для измерений и ресурсов регистра, связанных с признаками учета, платформа создает пару полей для хранения значения каждого ресурса отдельно по дебету и отдельно по кредиту проводки – КоличествоДт и КоличествоКт. А также для счетов, по которым ведется учет в разрезе субконто, платформа создает коллекции СубконтоДт и СубконтоКт.

Если обратиться к таблице 16.1, то при проведении приходной накладной счетом по дебету должен быть счет Товары (41), а счетом по кредиту – счет РасчетыСПоставщиками (60).

К счету мы обращаемся с помощью свойства глобального контекста `ПланыСчетов`. Оно предоставляет доступ ко всем планам счетов, созданным в конфигурации. Через точку от него мы указываем имя нужного нам плана счетов – `Основной`. А далее, тоже через точку, указываем имя предопределенного счета в этом плане счетов – `Товары`. Этот счет (и три других) мы создали в конфигураторе.

Так как количественный учет у нас ведется только для счета `Товары (41)`, то поле регистра `КоличествоДт` заполняется количеством товара из табличной части документа. Поле регистра `КоличествоКт` не заполняется, так как по счету кредита проводки (`РасчетыСПоставщиками`) количественный учет не ведется.

Теперь рассмотрим последнюю строку цикла, в которой присваивается значение субконто дебета.

Дело в том, что количество субконто на счете дебета и на счете кредита в каждой проводке будет различное, в зависимости от того, как определены счета в используемом плане счетов. Поэтому для каждой записи движения регистра бухгалтерии платформа хранит две коллекции значений: коллекцию субконто дебета и коллекцию субконто кредита. Каждая из них содержит ровно столько элементов, сколько видов субконто указано использовать для соответствующего счета (дебета или кредита) в плане счетов.

Обратиться к элементу коллекции можно, указав в квадратных скобках ссылку на соответствующий вид субконто (`ПланыВидовХарактеристик.ВидыСубконто.Материалы`).

Другой способ – в явном виде указать имя предопределенного вида субконто через точку от коллекции субконто дебета.

Другими словами, запись `Движение.СубконтоДт[ПланыВидовХарактеристик.ВидыСубконто.Материалы]` равносильна записи `Движение.СубконтоДт.Материалы`.

Коллекция регистра `СубконтоКт` не заполняется, так как по счету кредита проводки (`РасчетыСПоставщиками`) учет в разрезе субконто у нас не ведется.

В заключение отредактируем командный интерфейс формы документа, чтобы в панели навигации формы иметь возможность переходить к списку записей регистра `Управленческий`, связанному с документом.

Для этого откроем форму документа `ПриходнаяНакладная`. В левом верхнем окне перейдем на закладку `Командный интерфейс`. В разделе

Панель навигации раскроем группу Перейти и установим видимость для команды открытия регистра бухгалтерии Управленческий.

В режиме «1С:Предприятие»

Запустим «1С:Предприятие» в режиме отладки.

Платформа предупредит нас, что регистр бухгалтерии Управленческий и справочник Субконто не включены ни в одну подсистему. Проигнорируем это сообщение.

Откроем документ Приходная накладная № 1 и нажмем Провести.

Выполним команду перехода к регистру Управленческий и посмотрим, какие движения сформировал документ в регистре бухгалтерии (рис. 16.14, 16.15).

Движения в регистре Управленческий

Найти... Отменить поиск

Еще ▾

Период	↓	Регистратор	Н...	Счет Дт	Субконто1 Дт
07.07.2013 17:20:00		Приходная накладная 000000001...	1	41	Строчный трансформатор GoldStar (материал)
07.07.2013 17:20:00		Приходная накладная 000000001...	2	41	Строчный трансформатор Samsung (материал)
07.07.2013 17:20:00		Приходная накладная 000000001...	3	41	Транзистор Philips 2N2369 (материал)

Рис. 16.14. Движения документа «Приходная накладная № 1» в регистре бухгалтерии «Управленческий»

Движения в регистре Управленческий

Найти... Отменить поиск

Еще ▾

Субконто2 Дт	Количество Дт	Счет Кт	Субконто1 Кт	Субконто2 Кт	Количество Кт	Сумма
	10,000	60				2 700,00
	10,000	60				6 000,00
	10,000	60				30,00

Рис. 16.15. Движения документа «Приходная накладная № 1» в регистре бухгалтерии «Управленческий»

Обратите внимание: поскольку на счете 60 (РасчетыСПоставщиками) отсутствует аналитика и ведется только суммовой учет, в записях движений регистра Субконто1Кт, Субконто2Кт и КоличествоКт не указаны.

После этого перепроведем документ Приходная накладная № 2 и убедимся, что он тоже формирует правильные проводки по регистру бухгалтерии Управленческий.

Теперь перейдем к более сложной задаче: добавлению движений по регистру Управленческий в документ ОказаниеУслуги.

Доработка документа «Оказание услуги»

Сначала мы изменим процедуру проведения документа ОказаниеУслуги, а затем в режиме 1С:Предприятие перепроведем все эти документы, чтобы отработал новый, измененный нами алгоритм проведения документов Оказание услуги.

В режиме «Конфигуратор»

В отличие от документа ПриходнаяНакладная, который создавал всего одну бухгалтерскую проводку, документ ОказаниеУслуги будет создавать уже две.

Напомним, что бухгалтерия нашего ООО «На все руки мастер» не совсем похожа на настоящую, потому что для облегчения своей работы она учитывает только движения материалов. Услуги, которые оказывает организация, для нее как бы не существуют.

Поэтому документ ОказаниеУслуги будет формировать движения по регистру бухгалтерии только в той части, которая касается расхода материалов.

В окне редактирования объекта конфигурации Документ ОказаниеУслуги перейдем на закладку Движения. В списке регистров отметим, что документ будет создавать теперь движения и по регистру Управленческий. Перейдем на закладку Прочее и откроем модуль объекта. Откроем процедуру обработчика события ОбработкаПроведения.

Поскольку нас интересует только движение материалов, для внесения дополнений подойдет тело условия Если..., в котором мы формировали движения по регистрам ОстаткиМатериалов и СтоимостиМатериалов.

Добавим в конец условия, перед строкой КонецЕсли, движения по регистру бухгалтерии Управленческий (листинг 16.3).

В первой проводке мы указываем розничную сумму материала из документа и субконто дебета, поскольку на счете Дебиторская задолженность ведется учет в разрезе клиентов.

Листинг 16.3. Движения документа «ОказаниеУслуги» (фрагмент)

```
...
Если ВыборкаДетальныеЗаписи.ВидНоменклатуры =
    Перечисления.ВидыНоменклатуры.Материал Тогда

    // Регистр ОстаткиМатериалов Расход
    ...

    // Регистр СтоимостьМатериалов Расход
    ...

    // Регистр Управленческий
    // Первая проводка: Д 62(ДебиторскаяЗадолженность) – К 90 (Капитал) Розничная сумма
    Движение = Движения.Управленческий.Добавить();
    Движение.СчетДт = ПланыСчетов.Основной.ДебиторскаяЗадолженность;
    Движение.СчетКт = ПланыСчетов.Основной.Капитал;
    Движение.Период = Дата;
    Движение.Сумма = ВыборкаДетальныеЗаписи.СуммаВДокументе;
    Движение.СубконтоДт[ПланыВидовХарактеристик.ВидыСубконто.Клиенты] = Клиент;

    // Вторая проводка: Д 90 (Капитал) – К 41 (Товары) – себестоимость
    Движение = Движения.Управленческий.Добавить();
    Движение.СчетДт = ПланыСчетов.Основной.Капитал;
    Движение.СчетКт = ПланыСчетов.Основной.Товары;
    Движение.Период = Дата;
    Движение.Сумма = СтоимостьМатериала * ВыборкаДетальныеЗаписи.КоличествоВДокументе;
    Движение.КоличествоКт = ВыборкаДетальныеЗаписи.КоличествоВДокументе;
    Движение.СубконтоКт[ПланыВидовХарактеристик.ВидыСубконто.Материалы] =
        ВыборкаДетальныеЗаписи.Номенклатура;

КонецЕсли;
...
```

Во второй проводке мы указываем стоимость материала, количество и субконто кредита, поскольку на счете Товары ведется количественный учет в разрезе материалов.

Теперь в самом начале процедуры установим свойство Записывать регистра бухгалтерии в значение Истина для записи изменений регистров в базу данных (листинг 16.4).

Листинг 16.4. Запись движений регистров

Процедура ОбработкаПроведения(Отказ, Режим)

```
Движения.ОстаткиМатериалов.Записывать = Истина;
Движения.СтоимостьМатериалов.Записывать = Истина;
Движения.Продажи.Записывать = Истина;
Движения.Управленческий.Записывать = Истина;
...
```

В заключение отредактируем командный интерфейс формы документа, чтобы в панели навигации формы иметь возможность переходить к списку записей регистра Управленческий, связанному с документом. Для этого откроем форму документа ОказаниеУслуги. В левом верхнем окне перейдем на закладку Командный интерфейс. В разделе Панель навигации раскроем группу Перейти и установим видимость для команды открытия регистра бухгалтерии Управленческий.

В режиме «1С:Предприятие»

Запустим «1С:Предприятие» в режиме отладки, откроем документ Оказание услуги № 1 и нажмем Провести.

Выполним команду перехода к регистру Управленческий и посмотрим, какие движения сформировал документ в регистре бухгалтерии (рис. 16.16, 16.17).

Движения в регистре Управленческий

Найти... Отменить поиск

Еще ▾

Период	Регистратор	Н...	Счет Дт	Субконто1 Дт	Субконто2 Дт
10.07.2013 21:14:12	Оказание услуги 000000001 ...	1	62	Иванов Михаил Юрьевич	
10.07.2013 21:14:12	Оказание услуги 000000001 ...	2	90		

Рис. 16.16. Движения документа «Оказание услуги № 1» в регистре бухгалтерии «Управленческий»

Движения в регистре Управленческий

Найти... Отменить поиск

Еще ▾

Субконто2 Дт	Количество Дт	Счет Кт	Субконто1 Кт	Субконто2 Кт	Количество Кт	Сумма
		90				150,00
		41	Шланг резиновый (материал)		1,000	100,00

Рис. 16.17. Движения документа «Оказание услуги № 1» в регистре бухгалтерии «Управленческий»

После этого перепроведем остальные документы Оказание услуги.

Оборотно-сальдовая ведомость

Теперь нам только осталось создать отчет для бухгалтерии предприятия «На все руки мастер», и наше знакомство с использованием регистра бухгалтерии будет закончено.

Единственный отчет, которым пользуется бухгалтерия нашего предприятия, – это отчет Оборотно-сальдовая ведомость (рис. 16.18).

Оборотно-сальдовая ведомость						
Параметры: Период: 01.07.2013 - 31.07.2013						
Счет, Наименование	Сальдо нач дт	Сальдо нач кт	Оборот дт	Оборот кт	Сальдо кон дт	Сальдо кон кт
Итого			12 320,00	12 320,00	9 928,00	9 928,00
41, Товары			9 330,00	1 196,00	8 134,00	
60, Расчеты с поставщиками				9 330,00		9 330,00
62, Дебиторская задолженность			1 794,00		1 794,00	
90, Капитал				1 196,00		598,00
Итого			12 320,00	12 320,00	9 928,00	9 928,00

Рис. 16.18. Результат отчета

В режиме «Конфигуратор»

Для того чтобы его создать, откроем конфигуратор и добавим новый объект конфигурации Отчет с именем ОборотноСальдоваяВедомость. Создадим новую схему компоновки данных и добавим Набор данных – запрос. Откроем конструктор запроса.

Запрос для набора данных

Бухгалтерский отчет Оборотно-сальдовая ведомость представляет собой таблицу, в строках которой перечислены все имеющиеся в плане счетов счета, а в колонках – начальное сальдо, оборот и конечное сальдо по дебету и кредиту каждого счета.

Поэтому нам для построения такого отчета понадобятся две исходные таблицы:

- объектная (ссылочная) таблица плана счетов Основной;
- виртуальная таблица регистра бухгалтерии Управленческий.ОстаткиОбороты (рис. 16.19).




Рис. 16.19. Таблицы запроса

Из таблицы Основной мы выберем поле Ссылка, а из таблицы УправленческийОстаткиИОбороты возьмем следующие поля (рис. 16.20):

- СуммаНачальныйРазвернутыйОстатокДт,
- СуммаНачальныйРазвернутыйОстатокКт,
- СуммаОборотДт,
- СуммаОборотКт,
- СуммаКонечныйРазвернутыйОстатокДт,
- СуммаКонечныйРазвернутыйОстатокКт.




Рис. 16.20. Выбранные поля

Перейдем на закладку Связи и укажем, что из таблицы Основной мы будем выбирать все записи, а из таблицы регистра – только те, которые соответствуют условию связи (рис. 16.21).




Рис. 16.21. Условие связи таблиц

Затем на закладке Объединения/Псевдонимы зададим псевдонимы полей отчета: Счет, СальдоНачДт, СальдоНачКт, ОборотДт, ОборотКт, СальдоКонДт и СальдоКонКт (рис. 16.22).

Имя поля	Запрос 1
== Счет	== Основной.Ссылка
▣ СальдоНачДт	▣ УправленческийОстаткиИОбороты.СуммаНачальныйРазвернутыйОстатокДт
▣ СальдоНачКт	▣ УправленческийОстаткиИОбороты.СуммаНачальныйРазвернутыйОстатокКт
▣ ОборотДт	▣ УправленческийОстаткиИОбороты.СуммаОборотДт
▣ ОборотКт	▣ УправленческийОстаткиИОбороты.СуммаОборотКт
▣ СальдоКонДт	▣ УправленческийОстаткиИОбороты.СуммаКонечныйРазвернутыйОстатокДт
▣ СальдоКонКт	▣ УправленческийОстаткиИОбороты.СуммаКонечныйРазвернутыйОстатокКт

Рис. 16.22. Объединения/Псевдонимы

На этом создание запроса закончено, нажмем ОК.

Ресурсы

Перейдем на закладку Ресурсы и с помощью кнопки Добавить все ресурсы (>>) выберем все доступные ресурсы.

Параметры

Бухгалтерские отчеты, как правило, формируются для определенного периода: месяц, квартал, год и т.д. Поэтому на примере нашего отчета продемонстрируем использование стандартного периода для указания периода отчета.

На закладке Параметры добавим параметр с именем Период типа СтандартныйПериод, а для параметров НачалоПериода и КонецПериода укажем Выражение для расчета и запретим их редактирование пользователем (листинг 16.5).

Листинг 16.5. Выражение для расчета параметров «НачалоПериода» и «КонецПериода»

```
&Период.ДатаНачала
&Период.ДатаОкончания
```

Поскольку бухгалтерские отчеты всегда формируются за определенный период, для параметра Период в колонке Использование укажем значение Всегда. В результате в отчетной форме не будет доступен признак использования отчетного периода (флажок слева от параметра), и параметр будет использоваться всегда, независимо от желания пользователя.

Таким образом, параметры компоновки данных примут вид (рис. 16.23).




Рис. 16.23. Параметры схемы компоновки данных

Заметим, что даты начала и конца стандартного периода также содержат и время. Однако здесь, в отличие от параметров НачалоПериода и КонецПериода, начальная дата имеет время 00:00:00, а конечная дата – 23:59:59. Таким образом, последний день включается в отчет, и не нужно использовать функцию КонецПериода().

Настройки

В заключение перейдем на закладку Настройки и создадим структуру отчета. Добавим группировку, содержащую детальные записи. Затем на закладке Выбранные поля выберем поле Счет, затем реквизит этого поля Наименование (для этого раскроем поле Счет), а также все остальные поля для вывода в отчет и разместим их в следующем порядке (рис. 16.24):




Рис. 16.24. Поля и группировки отчета

На закладке **Другие настройки** укажем заголовок отчета – **Оборотно-сальдовая ведомость**. Для параметра **Расположение общих итогов по вертикали** укажем значение **Начало и конец**.

Затем на закладке **Параметры** выберем для параметра **Период** значение из списка стандартных периодов – **Этот месяц** (рис. 16.25).




Рис. 16.25. Создание быстрых настроек отчетного периода

Тем самым мы обеспечим, что при открытии формы отчета в настройке отчетного периода всегда будет указан текущий месяц. Причем даты начала и конца периода будут динамически меняться в зависимости от даты выполнения отчета. А также, нажав кнопку **Свойства элемента пользовательских настроек**, укажем, что параметр **Период** будет включен в состав пользовательских настроек, и эта настройка будет находиться непосредственно в отчетной форме (режим редактирования **Быстрый доступ**).

В заключение определим, в каких подсистемах будет отображаться наш отчет.

Закроем конструктор схемы компоновки данных и в окне редактирования объекта конфигурации **Отчет ОборотноСальдоваяВедомость** перейдем на закладку **Подсистемы**.

Отметим в списке подсистем конфигурации подсистему **Бухгалтерия**.

В режиме «1С:Предприятие»

Запустим «1С:Предприятие» в режиме отладки и посмотрим, как работает отчет.

В разделе Бухгалтерия откроем отчет и нажмем Сформировать (рис. 16.26).

Счет, Наименование	Сальдо нач дт	Сальдо нач кт	Оборот дт	Оборот кт	Сальдо кон дт	Сальдо кон кт
Итого			12 320,00	12 320,00	9 928,00	9 928,00
41, Товары			9 330,00	1 196,00	8 134,00	
60, Расчеты с поставщиками				9 330,00		9 330,00
62, Дебиторская задолженность			1 794,00		1 794,00	
90, Капитал			1 196,00	1 794,00		598,00
Итого			12 320,00	12 320,00	9 928,00	9 928,00

Рис. 16.26. Результат отчета

Мы видим, что стандартный период отчета задан по умолчанию — Этот месяц. Пользоваться стандартным периодом отчета очень удобно, когда пользователь регулярно выполняет отчет за определенный интервал времени. Тогда можно заранее установить в настройках нужный период, и пользователю не придется задавать его перед формированием отчета. Но при желании в режиме 1С:Предприятие пользователь может выбрать другой период из списка стандартных периодов.

Контрольные вопросы

- ✓ *Как использовать план видов характеристик для организации ведения бухгалтерского учета?*
- ✓ *Что такое субконто?*
- ✓ *Для чего предназначен объект конфигурации «План счетов»?*
- ✓ *Как создать план счетов?*
- ✓ *Для чего предназначен «Регистр бухгалтерии»?*
- ✓ *Как создать регистр бухгалтерии и настроить параметры учета?*
- ✓ *Как создать движения документа по регистру бухгалтерии средствами встроенного языка?*
- ✓ *Как создать отчет на основании данных из регистра бухгалтерии с помощью системы компоновки?*
- ✓ *Как задать стандартный период для выполнения отчета?*

ЗАНЯТИЕ 17

План видов расчета, регистр расчета

ПРОДОЛЖИТЕЛЬНОСТЬ

Ориентировочная продолжительность занятия – 1 час.

Зачем нужен план видов расчета и регистр расчета?	554
Зависимость по базовому периоду	555
Вытеснение по периоду действия	556
Что такое план видов расчета	557
Добавление плана видов расчета	560
В режиме «Конфигуратор»	560
Что такое регистр расчета	561
Периодичность	562
Вытеснение по периоду действия	564
Зависимость по базовому периоду	565
Зависимость по периоду действия	565
Зависимость по периоду регистрации	567
Добавление регистра расчета	568
В режиме «1С:Предприятие»	569
Контрольные вопросы	574

На этом занятии мы познакомимся с объектами конфигурации *План видов расчета* и *Регистр расчета* и узнаем об основных понятиях, используемых при создании сложных периодических расчетов.

В конце занятия мы создадим план видов расчета и регистр расчета, на основе которых на следующем занятии продемонстрируем работу механизмов периодических расчетов.

Зачем нужен план видов расчета и регистр расчета?

На этом занятии мы рассмотрим, какие возможности для автоматизации сложных периодических расчетов предоставляет система «1С:Предприятие».

Такие расчеты используются прежде всего при расчете заработной платы. Поэтому дальнейшее их рассмотрение мы будем строить на примере расчета заработной платы сотрудников, которые работают в нашем ООО «На все руки мастер».

В общем случае сумма заработной платы сотрудника складывается из множества частей (например, оплата по окладу, премии, штрафы, оплаты по больничному листу, разовые выплаты и т.д.). Каждая из этих частей рассчитывается по некоторому алгоритму, присущему только этой части.

Например, сумма штрафа может определяться просто фиксированной суммой, сумма премии может рассчитываться как процент от оклада, а сумма оплаты по окладу рассчитывается исходя из количества рабочих дней в месяце и количества дней, отработанных сотрудником. Поэтому для обозначения каждой такой части мы будем использовать термин *вид расчета*.

Алгоритм каждого вида расчета опирается в общем случае на две категории параметров: период, за который нужно получить конечные данные, и набор некоторых исходных данных, используемых при расчете.

Как правило, в реальной жизни различные виды расчета существуют не сами по себе, а оказывают некоторое влияние на другие виды расчета. Исходя из того, что вид расчета опирается на две различные категории параметров, такое влияние тоже имеет двойственный характер.

Зависимость по базовому периоду

Это может быть влияние на исходные данные, используемые при расчете.

В качестве примера можно привести начисление премии в виде процента от оплаты по окладу. При изменении оплаты по окладу размер премии тоже должен быть пересчитан, исходя из новой суммы начисленного оклада. Другими словами, сумма начисленного оклада является базой для расчета премии.

Причем поскольку оклад рассчитывается за некоторый период, при расчете премии нам интересно знать не значение оклада вообще, а сумму, начисленную в том периоде, который влияет на расчет премии. Такой период мы будем называть *базовым*, а подобную зависимость между видами расчета – *зависимостью по базовому периоду*.

В качестве примера рассмотрим начисление премии за апрель. Премия должна начисляться в размере 10 % от суммы, начисленной в качестве оплаты по окладу. Следовательно, необходимо проанализировать все записи о начислениях оплаты по окладу, которые попадают в интересующий нас базовый период, а именно апрель.

Допустим, общая сумма таких начислений составила 8 000 рублей – в этом случае премия должна быть начислена в размере 800 рублей (рис. 17.1).




Рис. 17.1. Зависимость премии от оклада по базовому периоду

Вытеснение по периоду действия

Это влияние может быть не на исходные данные, а на сам период, за который производится расчет.

В качестве примера можно привести расчет оплаты по окладу и невыход на работу. Предположим, мы начислили сотруднику оплату по окладу за март. В этом случае период действия такого расчета будет с 01.03.2013 по 31.03.2013.

После этого мы получили информацию от руководителя отдела, что, оказывается, сотрудник отсутствовал на работе с 1 по 10 марта по неизвестной причине. В этом случае нам нужно будет произвести расчет **Невыход** (в котором можно рассчитать какие-то удержания с сотрудника).

Но кроме этого необходимо будет пересчитать и оклад сотрудника, исходя из того, что фактический период действия расчета Оклад стал теперь с 11.03.2013 по 31.03.2013.

Такое влияние мы будем называть *вытеснением по периоду действия*.

В результате если за полный месяц работы сотруднику должно было быть начислено 9 300 рублей, то теперь, за фактический период работы, начисление составит 6 300 рублей (рис. 17.2).




Рис. 17.2. Запись расчета «Невыход» вытесняет запись расчета «Оклад» по периоду действия

Таким образом, исходя из двух видов взаимного влияния расчетов, можно сказать, что в общем случае с каждым видом расчета будет связано три периода: период действия, фактический период и базовый период.

- *Период действия* является «запрашиваемым». То есть, указывая период действия, мы говорим: «Мы хотели бы, чтобы результат действовал в этом периоде».
- *Фактический период* – это то, что получилось из периода действия после анализа всех периодов действия расчетов, которые вытесняют наш по периоду действия.
- *Базовый период* – это период, в котором мы анализируем результаты других расчетов, влияющих на наш по базовому периоду.

Как видите, взаимное влияние между видами расчетов может быть довольно разнообразным и, что самое сложное, многоуровневым. То есть один вид расчета может влиять на другой, который, в свою очередь, влияет на третий и т. д.

Очевидно, что в этой ситуации требуется некий универсальный механизм, позволяющий описать каждый из видов расчетов (его алгоритм, влияние на другие виды расчетов, зависимость от других видов расчетов), обеспечить хранение данных, полученных в результате этих расчетов, и контроль необходимости перерасчета результатов зависимых расчетов в случае изменения результатов первичных расчетов.

В системе «1С:Предприятие» такой универсальный механизм реализован при помощи планов видов расчета и регистров расчета.

И первым объектом конфигурации, с которым мы начнем знакомиться на этом занятии, будет План видов расчета.

Что такое план видов расчета

Объект конфигурации *План видов расчета* предназначен для описания структуры хранения информации о возможных видах расчетов. На основе объекта конфигурации План видов расчета платформа создает в базе данных таблицу, в которой будет храниться информация о том, какие существуют виды расчета и каковы взаимосвязи между ними.

Отличительной особенностью плана видов расчета является то, что пользователь в процессе работы может добавлять новые виды

расчета. Такая возможность делает механизм периодических расчетов более гибким и позволяет пользователю создавать собственные виды расчета, помимо тех, которые заданы разработчиком как предопределенные.

Объект конфигурации План видов расчета имеет свойство *Использует период действия*. С его помощью определяется, будут ли в этом плане находиться виды расчета, которые могут быть вытеснены по периоду действия.

Если это свойство установлено, то разработчик получает возможность указать для каждого вида расчета те виды, которые вытесняют его по периоду действия.

Следующим важным свойством объекта конфигурации План видов расчета является *Зависимость от базы*. Оно определяет, будут ли в этом плане находиться зависимые по базовому периоду виды расчета.

Если это свойство установлено, появляется возможность указать, в каком плане видов расчета будут находиться базовые виды расчета и, кроме этого, как будет определяться эта зависимость.

Существует возможность указать один из двух видов зависимости от базы: *Зависимость по периоду действия* и *Зависимость по периоду регистрации*. Оба вида этой зависимости подробно рассмотрены в разделе «Что такое регистр расчета» на стр. 561.

Еще одной важной особенностью плана видов расчета является возможность создания предопределенных видов расчета и описания их взаимного влияния. При этом в общем случае разработчик имеет возможность указать три категории видов расчета, влияющих на предопределенный вид расчета:

- *Базовые* – их результаты должны быть использованы при перерасчете этого вида расчета;
- *Вытесняющие* – вытесняют этот вид расчета по периоду действия;
- *Ведущие* – изменение их результатов должно приводить к необходимости перерасчета этого вида расчета.

Здравый смысл подсказывает, что все базовые виды расчета должны быть включены и в категорию ведущих. Кроме этого, ведущие виды расчета могут содержать и некоторые другие виды, косвенно влияющие на данный вид расчета.

Например, мы имеем три вида расчета: Невыход, Оклад и Премия. Невыход вытесняет Оклад по периоду действия, а Премия зависит от оклада по базовому периоду.

В этом случае для премии следует указать базовым видом расчета оклад, а ведущими – оклад и невыход, поскольку изменение результата расчета невыхода приведет к изменению результата оклада, что, в свою очередь, должно привести к изменению результата премии (рис. 17.3).



Рис. 17.3. Взаимное влияние видов расчетов

УЗНАЙ БОЛЬШЕ!

О структуре объектов встроенного языка, предназначенных для работы с планом видов расчета, можно прочитать в разделе «Краткий справочник разработчика. Планы видов расчета» на стр. 922.

Добавление плана видов расчета

Приступим теперь к созданию плана видов расчета ОсновныеНачисления, который будет использоваться в нашей конфигурации.

В режиме «Конфигуратор»

Откроем конфигуратор и создадим новый объект конфигурации План видов расчета.

Зададим его имя – ОсновныеНачисления, а также зададим Представление списка как Виды расчетов.

На закладке Подсистемы укажем, что план видов расчета будет отображаться в подсистеме РасчетЗарплаты.

На закладке Расчет укажем, что он будет использовать период действия и зависеть от базы по периоду действия.

В качестве базового плана видов расчета укажем его самого, поскольку все наши виды расчетов будут храниться в единственном плане видов расчета (рис. 17.4).




Рис. 17.4. Определим использование периода действия, зависимость от базы и базовые планы видов расчета

Перейдем на закладку Прочее и зададим предопределенные виды расчета.

Как и в случае с бухгалтерией, расчеты в нашем ООО «На все руки мастер» будут скромные, поэтому мы создадим всего три элемента (рис. 17.5):

- Невыход – с именем и наименованием Невыход и кодом Невыход;
- Оклад – с именем, кодом и наименованием Оклад, с вытесняющим его видом расчета Невыход и ведущим видом расчета Невыход;
- Премия – с именем, кодом и наименованием Премия, с базовым видом расчета Оклад и ведущими видами расчета Невыход и Оклад.




Рис. 17.5. Предопределенные виды расчета для плана видов расчета «ОсновныеНачисления»

Теперь мы перейдем к рассмотрению второго объекта, используемого при реализации механизмов сложных периодических расчетов, – регистра расчета.

Что такое регистр расчета

Объект конфигурации *Регистр расчета* предназначен для описания структуры накопления данных, являющихся результатами расчетов. На основе объекта конфигурации **Регистр расчета** платформа создает в базе данных таблицы, в которых будут накапливаться данные, формируемые различными объектами базы данных.

Отличительной особенностью регистра расчета является то, что он не предназначен для интерактивного редактирования пользователем. Разработчик может при необходимости предоставить пользователю

возможность редактировать регистр расчета, но предназначение регистра расчета заключается в том, чтобы его модификация производилась на основе алгоритмов работы объектов базы данных, а не в результате непосредственных действий пользователя.

Как и другие регистры, регистр расчета имеет ресурсы, в которых хранит числовые данные; имеет измерения, в разрезе которых можно получать значения ресурсов регистра; имеет реквизиты, которые характеризуют каждую запись регистра расчета.

Отличительными же особенностями регистра расчета является его периодичность, возможность использования механизмов вытеснения по периоду действия и зависимости по базовому периоду, а также связь с планом видов расчета. Рассмотрим все эти особенности по порядку.

Периодичность

Периодичность регистра расчета может быть определена одним из следующих значений:

- День,
- Месяц,
- Квартал,
- Год.

Периодичность регистра расчета определяет промежуток времени, к которому будет относиться каждая запись регистра.

Если указана периодичность **День**, то каждая запись регистра будет относиться к какому-либо дню; если периодичность – **Месяц**, то к какому-либо месяцу и т. д.

Для указания факта принадлежности записи к какому-либо периоду регистр имеет служебный реквизит **ПериодРегистрации** типа **Дата**. При записи данных в регистр платформа всегда приводит значение этого реквизита к началу того периода, в который он попадает.

Например, если в регистр расчета с периодичностью **месяц** записать данные, где **ПериодРегистрации** задан как **08.04.2004**, то регистр сохранит эти данные со значением поля **ПериодРегистрации** **01.04.2004** (рис. 17.6).

Если в этой же ситуации периодичность регистра будет **год**, сохраненное значение периода регистрации будет **01.01.2004** (рис. 17.7).



Рис. 17.6. Запись данных из документа в регистр расчета видов расчета




Рис. 17.7. Запись данных из документа в регистр расчета видов расчета

Вытеснение по периоду действия

Следующей важной особенностью регистра расчета является возможность использования механизма вытеснения одних записей другими по периоду действия.

При этом для каждой записи регистр расчета формирует фактический период действия, который является в общем случае совокупностью нескольких периодов, расположенных внутри периода действия (рис. 17.8).




Рис. 17.8. Запись расчета «Невыход» вытесняет запись расчета «Оклад» по периоду действия

Если рассмотреть структуру записей таблиц регистра расчета, то после внесения записи о начислении по окладу таблицы регистра будут выглядеть следующим образом (таблицы 17.1, 17.2).

Таблица 17.1. Таблица регистра расчета

...	Начало периода действия	Конец периода действия	Вид расчета	...
...	...	...	...	...
...	01.04.2013 00:00:00	30.04.2013 23:59:59	Оклад	...
...	...	...	...	...

Таблица 17.2. Таблица фактического периода действия

...	Начало периода действия	Конец периода действия	Вид расчета	...
...	...	...	...	...
...	01.04.2013 00:00:00	30.04.2013 23:59:59	Оклад	...
...	...	...	...	...

После добавления в регистр записи вида расчета **Невыход**, который вытесняет вид расчета **Оклад** по периоду действия, записи о начислении по окладу примут следующий вид (таблицы 17.3, 17.4).

Таблица 17.3. Таблица регистра расчета

...	Начало периода действия	Конец периода действия	Вид расчета	...
...	...	...	...	...
...	01.04.2013 00:00:00	30.04.2013 23:59:59	Оклад	...
...	04.04.2013 00:00:00	10.04.2013 23:59:59	Невыход	...
...	...	...	...	...

Таблица 17.4. Таблица фактического периода действия

...	Начало периода действия	Конец периода действия	Вид расчета	...
...	...	...	...	...
...	01.04.2013 00:00:00	03.04.2013 23:59:59	Оклад	...
...	11.04.2013 00:00:00	31.04.2013 23:59:59	Оклад	...
...	...	...	...	...

Зависимость по базовому периоду

Другим механизмом, который поддерживает регистр расчета, является зависимость записей по базовому периоду. Этот механизм позволяет основывать расчет зависимых (вторичных) записей регистра на данных, полученных в результате расчета первичных записей.

Регистр расчета может поддерживать два вида зависимости от базы: зависимость по периоду действия и зависимость по периоду регистрации.

Зависимость по периоду действия

Зависимость по периоду действия означает, что при анализе базовых записей будут выбираться те, для которых найдено пересечение их фактического периода действия и указанного базового периода.

Например, в начале апреля производится расчет зарплаты за март. Премия за март должна быть начислена исходя из оплаты по окладу за март. В этом случае, как правило, используется зависимость по периоду действия (рис. 17.9).

Таблица регистра расчета

Регистратор	Номер строки	Начало периода действия	Конец периода действия	Начало базового периода	Конец базового периода	Вид расчета	Результат	...
...	...	...	...	...	...	...	...	...
● (док.3)	3	01.03.2004 00:00:00	31.03.2004 23:59:59	01.03.2004 00:00:00	31.03.2004 23:59:59	● (премия)	3 000 *X	...
...	...	...	...	...	...	...	...	...
● (док.2)	5	01.03.2004 00:00:00	31.03.2004 23:59:59	...	...	● (оклад)	3 000	...
...	...	...	...	...	...	...	...	...
● (док.1)	2	01.04.2004 00:00:00	30.04.2004 23:59:59	...	...	● (оклад)	5 000	...
...	...	...	...	...	...	...	...	...

Таблица фактического периода действия

Регистратор	Номер строки	Начало периода действия	Конец периода действия	...
...	...	...	...	...
● (док.3)	3	01.03.2004 00:00:00	25.03.2004 23:59:59	...
...	...	...	...	...
● (док.2)	5	01.03.2004 00:00:00	31.03.2004 23:59:59	...
...	...	...	...	...
● (док.1)	2	02.04.2004 00:00:00	03.04.2004 23:59:59	...
● (док.1)	2	05.04.2004 00:00:00	21.04.2004 23:59:59	...
● (док.1)	2	23.04.2004 00:00:00	30.04.2004 23:59:59	...
...	...	...	...	...

Рис. 17.9. Зависимость по периоду действия

Следует сделать два замечания к приведенному рисунку.

Поля Начало базового периода и Конец базового периода имеют смысл только для записей тех видов расчета, для которых определена зависимость по базовому периоду (в нашем случае для записи расчета премии).

Значение базы, которая будет получена от конкретной влияющей записи, в общем случае не равно результату, который содержит эта запись. База будет рассчитана пропорционально тому, какую часть от фактического интервала влияющей записи составляет перекрывающийся с указанным базовым периодом участок. При этом будут использованы данные графика, связанного с записью.

Зависимость по периоду регистрации

Зависимость по периоду регистрации означает, что при анализе базовых записей будут выбираться те, которые попадают в указанный базовый период значением своего поля Период регистрации.

В качестве примера можно привести расчет штрафов при начислении зарплаты за март. В качестве базы для расчета суммы штрафов должны браться записи о прогулах, зарегистрированные в марте месяце (это могут быть как записи о мартовских прогулах, так и записи о прогулах в феврале). В этом случае, как правило, используется зависимость по периоду регистрации (рис. 17.10).

Таблица регистра расчета
(периодичность регистр - «Месяц»)

Период регистрации	Начало периода действия	Конец периода действия	Начало базового периода	Конец базового периода	Вид расчета	Результат
...	...	...	...	...	...	...
01.04.2004 00:00:00	01.03.2004 00:00:00	31.03.2004 23:59:59	01.03.2004 00:00:00	31.03.2004 23:59:59	○ (штраф)	2*Х
...	...	...	...	...	...	...
01.03.2004 00:00:00	26.02.2004 00:00:00	27.02.2004 23:59:59	...	...	○ (прогул)	2
...	...	...	...	...	...	...
01.02.2004 00:00:00	07.02.2004 00:00:00	10.02.2004 00:00:00	...	...	○ (прогул)	4
...	...	...	...	...	...	...

Рис. 17.10. Зависимость по периоду регистрации

Важной заключительной особенностью регистра расчета является его связь с планом видов расчета. Именно на основе этой связи работают механизмы вытеснения по периоду действия и зависимости по базовому периоду, поскольку в плане видов расчета описано взаимное влияние видов расчета друг на друга.

У регистра расчета могут существовать подчиненные объекты *Перерасчет*. Они предназначены для регистрации фактов появления в регистре записей, влияющих на результат расчета уже существующих записей регистра. Объект конфигурации Перерасчет может иметь несколько измерений, каждое из которых устанавливает связь между измерениями данного регистра расчета и влияющих регистров расчета. В частном случае это может быть один и тот же регистр.

В таблице, созданной в базе данных на основе объекта конфигурации Перерасчет, платформа хранит информацию о том, какие записи регистра подлежат перерасчету. Таблицы перерасчета

заполняются автоматически как на основании записей регистров расчета, затронутых ведущими видами расчета, так и на основании записей регистра расчета, для которых изменился фактический период действия. Исходя из этой информации, разработчик может принимать решение о необходимости перерасчета записей регистра.

Последним замечанием, которое следует сделать, говоря о регистре расчета, является возможность установки связи регистра расчета с графиком. Такой график должен представлять собой регистр сведений (непериодический, с обязательным измерением типа Дата и ресурсом типа Число), в котором содержится временная схема исходных данных, участвующих в расчетах. Измерениями этого графика могут быть, например, график работы (ссылка на справочник) и дата, а ресурсом – количество рабочих часов в этой дате. В этом случае можно будет связать запись регистра расчета с каким-либо конкретным графиком работы (указав в качестве реквизита записи ссылку на справочник ВидыГрафиковРаботы) и в дальнейшем средствами встроенного языка получать информацию о количестве рабочих часов в периоде действия, фактическом периоде действия или периоде регистрации этой записи.

УЗНАЙ БОЛЬШЕ!

О структуре объектов встроенного языка, предназначенных для работы с регистром расчета, можно прочитать в разделе «Краткий справочник разработчика. Регистры расчета» на стр. 938.

Добавление регистра расчета

Прежде чем мы начнем создавать объект конфигурации Регистр расчета Начисления, нам потребуется создать два дополнительных объекта конфигурации:

- регистр сведений ГрафикиРаботы,
- справочник ВидыГрафиковРаботы.

Справочник понадобится нам для хранения информации о том, какие графики работы существуют в ООО «На все руки мастер», а регистр сведений – для указания того, какие дни в месяце являются рабочими, поскольку сумма оплаты по окладу будет рассчитываться исходя из того, сколько дней отработал сотрудник в расчетном месяце.

В режиме «Конфигуратор»

Откроем конфигуратор и создадим новый объект конфигурации Справочник с именем ВидыГрафиковРаботы.

На закладке Подсистемы укажем, что справочник будет отображаться в подсистеме РасчетЗарплаты.

На закладке Прочее создадим для справочника два predetermined графика работы (рис. 17.11) – ГрафикАдминистрации и ГрафикМастеров.




Рис. 17.11. Предопределенные графики работы

После этого создадим объект конфигурации Регистр сведений с именем ГрафикиРаботы.

Этот регистр будет иметь два измерения:

- ГрафикРаботы, тип СправочникСсылка.ВидыГрафиковРаботы;
- Дата, тип Дата.

Затем создадим единственный ресурс регистра – Значение с типом Число, длиной 1. На закладке Подсистемы укажем, что регистр сведений будет отображаться в подсистеме РасчетЗарплаты. Теперь заполним регистр сведений ГрафикиРаботы данными о рабочих днях июля графика мастеров.

В режиме «1С:Предприятие»

Запустим «1С:Предприятие» в режиме отладки и в разделе Расчет зарплаты выполним команду Графики работы. Поочередно создадим 31 запись в регистре.

Чтобы проще выполнить эту довольно однообразную работу, воспользуйтесь возможностью добавления элементов в регистр

копированием текущего элемента (команда Еще ► Скопировать (F9)). В качестве измерения ГрафикРаботы нашего регистра выберем predetermined элемент График мастеров справочника ВидыГрафиковРаботы. В качестве ресурса Значение у рабочих дней проставим 1, а у выходных – 0 (рис. 17.12).




Рис. 17.12. Записи регистра «Графики работы»

Теперь все готово для создания регистра расчета.

В режиме «Конфигуратор»

Добавим новый объект конфигурации Регистр расчета с именем Начисления. Зададим Расширенное представление списка как Движения в регистре Начисления. В качестве плана видов расчета, используемого регистром, выберем ОсновныеНачисления.

Установим, что регистр будет использовать период действия, график будет задаваться в регистре сведений ГрафикиРаботы, значение графика будет находиться в ресурсе Значение, а дата графика – в измерении Дата.

Укажем, что регистр расчета будет использовать базовый период и периодичность регистра будет Месяц (рис. 17.13).




Рис. 17.13. Окно редактирования регистра расчета «Начисления»

На закладке Подсистемы укажем, что регистр расчета будет отображаться в подсистеме РасчетЗарплаты.

Затем перейдем на закладку Данные и создадим (рис. 17.14):

- измерение Сотрудник, тип СправочникСсылка.Сотрудники, Базовое;
- ресурс Результат, тип Число, длина 15, точность 2,

- реквизит ГрафикРаботы, тип СправочникСсылка.ВидыГрафиковРаботы, в разделе свойств Данные зададим свойство Связь с графиком по измерению ГрафикРаботы;
- реквизит ИсходныеДанные, тип Число, длина 15, точность 2.

Реквизит ГрафикРаботы мы будем использовать для того, чтобы связать запись регистра с используемым графиком работы, а реквизит ИсходныеДанные – чтобы хранить в нем данные, которые могут понадобиться при расчете или перерасчете (в нашем примере это будет расчет оклада).




Рис. 17.14. Измерения, ресурсы и реквизиты регистра расчета

Теперь перейдем на закладку Перерасчеты. Создадим объект конфигурации Перерасчет, который так и назовем – Перерасчет.

У него будет единственное измерение – Сотрудник, для которого в разделе Связь мы укажем:

- измерение регистра – Сотрудник;

- данные ведущих регистров – выберем то же самое измерение Сотрудник регистра расчета Начисления (рис. 17.15).




Рис. 17.15. Перерасчеты регистра

В заключение отредактируем командный интерфейс, чтобы в разделе Расчет зарплаты была доступна команда для просмотра записей регистра расчета.

Для этого в дереве объектов конфигурации выделим подсистему РасчетЗарплаты, вызовем контекстное меню и выполним команду Открыть командный интерфейс.

В открывшемся окне командного интерфейса подсистемы в группе Панель навигации.Обычное включим видимость у команды Начисления.

На этом создание объекта конфигурации Регистр расчета Начисления завершено.

Контрольные вопросы

- ✓ *Что такое сложные периодические расчеты?*
- ✓ *Что такое вид расчета, база?*
- ✓ *Какая разница между базовым периодом, фактическим периодом и периодом действия?*
- ✓ *Что такое зависимость по базовому периоду?*
- ✓ *Что такое вытеснение по периоду действия?*
- ✓ *Для чего предназначен объект конфигурации «План видов расчета»?*
- ✓ *Каковы основные свойства плана видов расчета?*
- ✓ *Какая разница между базовыми, вытесняющими и ведущими видами расчетов?*
- ✓ *Как создать план видов расчета?*
- ✓ *Что такое объект конфигурации «Регистр расчета»?*
- ✓ *Каковы отличительные особенности регистра расчета?*
- ✓ *Что такое график времени?*
- ✓ *Что такое перерасчет?*
- ✓ *По какому принципу формируются записи перерасчета?*
- ✓ *Как создать регистр расчета?*

ЗАНЯТИЕ 18

Использование регистра расчета

ПРОДОЛЖИТЕЛЬНОСТЬ

Ориентировочная продолжительность занятия – 3 часа 40 минут.

Добавление документа о начислениях	576
В режиме «Конфигуратор»	576
В режиме «1С:Предприятие»	580
Иллюстрация механизмов вытеснения и зависимости от базы	582
Отчет по перерасчетам	582
Зависимость по базовому периоду	583
Вытеснение по периоду действия	584
Процедура расчета записей регистра расчета	586
Отчет о начислениях сотрудникам	594
В режиме «Конфигуратор»	595
Запрос для набора данных	595
Ресурсы	596
Настройки	596
В режиме «1С:Предприятие»	597
Перерасчет	598
Диаграмма Ганта	603
Контрольные вопросы	614

Теперь у нас все готово для того, чтобы начать разработку системы расчета заработной платы ООО «На все руки мастер».

В этой главе мы создадим документ, с помощью которого будут выполняться различные виды начислений; посмотрим, как и когда платформа формирует записи перерасчета; увидим, как работают механизмы вытеснения по периоду действия и зависимости по базовому периоду.

Кроме этого, мы создадим отчет, показывающий начисления сотрудникам ООО «На все руки мастер», и сделаем так, чтобы данные расчетов можно было поддерживать в актуальном состоянии.

В заключение мы познакомимся с новым элементом формы – *Диаграмма Ганта* – и с его помощью наглядно проиллюстрируем работу некоторых механизмов расчета.

Добавление документа о начислениях

Для того чтобы иметь возможность регистрировать в базе данных начисления, производимые сотрудникам ООО «На все руки мастер», нам понадобится специальный документ.

В режиме «Конфигуратор»

Откроем конфигуратор и добавим новый объект конфигурации Документ. Назовем его НачисленияСотрудникам. Зададим представление объекта как Начисление сотрудникам. На закладке Нумерация установим:

- Тип номера – Число,
- Длина номера – 5 (рис. 18.1).

На закладке Подсистемы укажем, что документ будет отображаться в подсистеме РасчетЗарплаты.

На закладке Данные укажем, что этот документ будет иметь табличную часть Начисления, содержащую следующие реквизиты:

- Сотрудник, тип СправочникСсылка.Сотрудники;
- ГрафикРаботы, тип СправочникСсылка.ВидыГрафиковРаботы;
- ДатаНачала, тип Дата;
- ДатаОкончания, тип Дата;
- ВидРасчета, тип ПланВидовРасчетаСсылка.ОсновныеНачисления;
- Начислено, Число, длина 15, точность 2.




Рис. 18.1. Нумерация документа

Реквизиты `ДатаНачала` и `ДатаОкончания` понадобятся нам для того, чтобы задавать период, в котором должна действовать запись расчета.

На закладке `Движения` запретим оперативное проведение документа.

Отметим, что документ будет создавать движения по регистру расчета `Начисления`, и запустим конструктор движений (рис. 18.2).




Рис. 18.2. Движения документа

В окне конструктора выберем табличную часть Начисления и нажмем Заполнить выражения.

Для реквизитов ПериодДействияКонец и БазовыйПериодКонец укажем выражение КонецДня(ТекСтрокаНачисления.ДатаОкончания).

Для поля ПериодРегистрации укажем выражение Дата.

Реквизиту ИсходныеДанные поставим в соответствие реквизит табличной части Начислено, а для ресурса Результат оставим пустое выражение, так как мы будем его потом рассчитывать (рис. 18.3).




Рис. 18.3. Движения документа «НачисленияСотрудникам» по регистру расчета

Нажмем ОК и посмотрим текст обработчика, созданный конструктором (листинг 18.1).

Листинг 18.1. Текст обработчика, созданный конструктором движений

Процедура ОбработкаПроведения(Отказ, Режим)

```
//({_КОНСТРУКТОР_ДВИЖЕНИЙ_РЕГИСТРОВ
// Данный фрагмент построен конструктором
// При повторном использовании конструктора
// внесенные вручную изменения будут утеряны!!!
```



```
Движения.Начисления.Записывать = Истина;
```

```
// Регистр Начисления  
Для Каждого ТекСтрокаНачисления Из Начисления Цикл
```

```
    Движение = Движения.Начисления.Добавить();  
    Движение.Сторно = Ложь;  
    Движение.ВидРасчета = ТекСтрокаНачисления.ВидРасчета;  
    Движение.ПериодДействияНачало = ТекСтрокаНачисления.ДатаНачала;  
    Движение.ПериодДействияКонец = КонецДня(ТекСтрокаНачисления.ДатаОкончания);  
    Движение.ПериодРегистрации = Дата;  
    Движение.БазовыйПериодНачало = ТекСтрокаНачисления.ДатаНачала;  
    Движение.БазовыйПериодКонец = КонецДня(ТекСтрокаНачисления.ДатаОкончания);  
    Движение.Сотрудник = ТекСтрокаНачисления.Сотрудник;  
    Движение.ГрафикРаботы = ТекСтрокаНачисления.ГрафикРаботы;  
    Движение.ИсходныеДанные = ТекСтрокаНачисления.Начислено;
```

```
КонецЦикла;  
//}}_КОНСТРУКТОР_ДВИЖЕНИЙ_РЕГИСТРОВ
```

КонецПроцедуры

Прокомментируем этот код.

В целом в нем нет ничего принципиально нового по сравнению с прежними обработчиками проведения документов, создающими движения регистров.

Напомним лишь, что Начисления – имя регистра расчета, и в выражениях Движения.Начисления.Записывать() и Движения.Начисления.Добавить() мы обращаемся к методам набора записей этого регистра.

В строке Для Каждого ТекСтрокаНачисления Из Начисления Цикл мы обращаемся по имени (Начисления) к табличной части нашего документа и организуем цикл обхода этой табличной части.

В цикле мы добавляем к движениям новую запись и присваиваем ее полям значения из табличной части документа.

Для присвоения значений полям ПериодДействияКонец, БазовыйПериодКонец мы используем функцию КонецДня(). Аналогичным образом мы поступали при создании отчетов, чтобы последний день периода попал в отчет.

В заключение отредактируем командный интерфейс, чтобы в разделе Расчет зарплаты была доступна команда создания новых документов.

Для этого в дереве объектов конфигурации выделим подсистему РасчетЗарплаты, вызовем контекстное меню и выполним команду Открыть командный интерфейс.

В группе **Панель действий.Создать** включим видимость у команды **Начисление сотрудникам: создать** (рис. 18.4).




Рис. 18.4. Редактирование командного интерфейса подсистемы

В режиме «1С:Предприятие»

Запустим «1С:Предприятие» в режиме отладки и посмотрим, как работает наш документ.

В разделе **Расчет зарплаты** выполним команду **Начисление сотрудникам** и начислим оклад за июль всем сотрудникам ООО «На все руки мастер» (рис. 18.5).




Рис. 18.5. Документ «Начисления сотрудникам № 1»

Проведем документ и посмотрим, какие движения он сформировал в регистре Начисления (рис. 18.6, 18.7).

Период регистрации	Регистратор	Н.	Вид расчета	Сторно	Сотрудник	Результат	График работы	Исходные данные
01.07.2013 0:00:00	Начисление с...	1	Оклад		Гусаков Николай Дмитриевич		График мастеров	10 000,00
01.07.2013 0:00:00	Начисление с...	2	Оклад		Деловой Иван Сергеевич		График мастеров	8 000,00
01.07.2013 0:00:00	Начисление с...	3	Оклад		Симонов Валерий Михайлов...		График мастеров	3 000,00

Рис. 18.6. Движения документа «Начисление сотрудникам № 1» в регистре расчета «Начисления»

Период действия	Дата начала периода действия	Дата окончания периода действия	Дата начала базового периода	Дата окончания базового периода
01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59
01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59
01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59

Рис. 18.7. Движения документа «Начисление сотрудникам № 1» в регистре расчета «Начисления»

Обратите внимание на то, что платформа привела период регистрации каждой записи к началу периода регистра расчета – началу месяца (в обработчике проведения мы указывали значение даты документа – 24.07.2013).

Кроме этого заметьте, что в каждой записи мы сохранили в реквизите ИсходныеДанные размер оклада сотрудника, введенный в документе, чтобы в дальнейшем рассчитать сумму оплаты по окладу.

Для дальнейшего изучения работы регистра расчета нам понадобится служебный отчет, с помощью которого мы сможем посмотреть содержимое записей перерасчета.

Иллюстрация механизмов вытеснения и зависимости от базы

Отчет по перерасчетам

В режиме «Конфигуратор»

Создадим новый объект конфигурации Отчет. Назовем его Перерасчет. Создадим основную схему компоновки данных, добавим источник данных – запрос и откроем конструктор запроса.

В списке База данных раскроем ветвь Перерасчеты и из виртуальной таблицы перерасчета Начисления.Перерасчет выберем все поля:

- ОбъектПерерасчета,
- ВидРасчета,
- Сотрудник (рис. 18.8).




Рис. 18.8. Поля и таблицы запроса

На этом создание запроса закончено, нажмем ОК.

Перейдем на закладку Настройки и добавим группировку детальных записей. На закладке Выбранные поля выберем для вывода в отчет поля ОбъектПерерасчета, ВидРасчета и Сотрудник.

На этом создание схемы компоновки данных закончено, закроем ее.

В окне редактирования объекта конфигурации Отчет Перерасчет на закладке Подсистемы укажем, что отчет будет принадлежать подсистеме РасчетЗарплаты.

Зависимость по базовому периоду

В режиме «1С:Предприятие»

Если сейчас мы выполним отчет в режиме 1С:Предприятие, то мы увидим, что ни один перерасчет еще не выполнялся.

Поэтому создадим новый документ Начисление сотрудникам № 2, в котором начислим премию за июль Гусакову и Деловому (рис. 18.9).

N	Сотрудник	График работы	Дата начала	Дата окончания	Вид расчета	Начислено
1	Гусakov Николай Дмитриевич	График мастеров	01.07.2013	31.07.2013	Премия	
2	Деловой Иван Сергеевич	График мастеров	01.07.2013	31.07.2013	Премия	

Рис. 18.9. Документ «Начисления сотрудникам № 2»

Этим документом мы зафиксируем тот факт, что сотрудникам Гусакову и Деловому нужно начислить премию по итогам работы за июль. Поскольку размер премии нам неизвестен (он будет рассчитываться по некоторому алгоритму), поле Начислено мы оставляем пустым. Нажмем Провести и закрыть.

Теперь снова откроем документ Начисление сотрудникам № 1 и изменим оклад Гусакова с 10000 на 7000. Нажмем Провести и закрыть.

Сформируем отчет Перерасчет (рис. 18.10).

Регистратор	Вид расчета	Сотрудник
Начисление сотрудникам 2 от 24.07.2013 20:07:48	Премия	Гусakov Николай Дмитриевич
Начисление сотрудникам 2 от 24.07.2013 20:07:48	Премия	Деловой Иван Сергеевич

Рис. 18.10. Отчет «Перерасчет»

Как видите, отчет теперь содержит какие-то данные.

В самом деле вид расчета Премия зависит у нас по базовому периоду от вида расчета Оклад. Как только мы изменили существовавшие в регистре записи по виду расчета Оклад, платформа сразу же сформировала набор записей перерасчета, которые должны быть рассчитаны заново, так как изменилась их база.

Вы можете спросить: почему в перерасчет попали записи как про Делового, так и про Гусакова, хотя оклад мы меняли только Гусакову?

Дело в том, что платформа не отслеживает конкретные изменения, которые пользователь внес в записи документа. Она отслеживает лишь факт изменения набора записей регистра расчета в результате проведения (перепроведения) документа.

Поэтому в набор записей перерасчета она включает информацию обо ВСЕХ записях регистра, значение ресурсов которых МОЖЕТ измениться в результате перепроведения документа, создавшего базовые записи регистра.

Перепроведем документ Начисления сотрудникам № 2 (которым мы начисляли премию) и сформируем отчет Перерасчет.

Он снова не содержит никаких данных – система отметила тот факт, что мы «пересчитали» зависимые записи, и очистила таблицу перерасчета.

На этом примере мы с вами познакомились с работой механизма поддержки зависимости по базовому периоду у регистра расчета.

Вытеснение по периоду действия

В режиме «1С:Предприятие»

Теперь посмотрим, как работает механизм вытеснения по периоду действия.

Для этого нам понадобится создать документ Начисления сотрудникам № 3 (рис. 18.11).

Этим документом мы зафиксируем тот факт, что Гусаков не выходил на работу с 1 по 10 июля.

Очевидно, что в этом случае потребуется пересчитать его оплату по окладу и как следствие начисленную премию.

Начисление сотрудникам (создание) *

Провести и закрыть Записать Провести Еще ▾

Номер:

Дата:

Добавить Еще ▾

N	Сотрудник	График работы	Дата начала	Дата окончания	Вид расчета	Начислено
1	Гусаков Николай Дмитриевич	График мастеров	01.07.2013	10.07.2013	Невыход	

Рис. 18.11. Документ «Начисления сотрудникам № 3»

Нажмем Провести и закрыть и затем сформируем отчет Перерасчет (рис. 18.12).

☆ Перерасчет

Сформировать Выбрать вариант... Еще ▾

Регистратор	Вид расчета	Сотрудник
Начисление сотрудникам 2 от 24.07.2013 20:07:48	Премия	Гусаков Николай Дмитриевич
Начисление сотрудникам 1 от 24.07.2013 19:28:40	Оклад	Гусаков Николай Дмитриевич

Рис. 18.12. Отчет «Перерасчет»

Как вы видите, в перерасчет попала запись о начислении оклада Гусакову. Это явилось результатом работы механизма вытеснения по периоду действия, ведь вид расчета Невыход вытесняет у нас вид расчета Оклад.

Обратите внимание, что в перерасчет попала и запись о начислении премии Гусакову.

Если вы помните, при создании предопределенных видов расчета мы указали, что результат вида расчета Премия будет зависеть от изменения результата вида расчета Невыход. Эта зависимость косвенная, но поскольку явно указали такую зависимость, платформа ее отследила.

Перепроведем документы Начисления сотрудникам № 1 и № 2 и убедимся, что таблица перерасчета очистилась.

Процедура расчета записей регистра расчета

В режиме «Конфигуратор»

До сих пор мы с вами просто заносили в регистр расчета Начисления записи о том, что необходимо выполнить какой-либо вид расчета. Но каким именно образом получать эти результаты, мы не говорили.

Теперь настало время описать алгоритмы формирования различных видов расчетов.

Поскольку эти алгоритмы нужно будет использовать не только в документе Начисление сотруднику, удобнее всего будет разместить их в отдельном общем модуле.

Откроем в конфигураторе текст обработчика проведения документа НачисленияСотрудникам и добавим в него после завершения создания движений в регистре Начисления вызов процедуры РассчитатьНачисления() из общего модуля ПроведениеРасчетов (листинг 18.2).

Листинг 18.2. Обработчик проведения документа «НачисленияСотрудникам»

Процедура ОбработкаПроведения(Отказ, Режим)

```
....
КонецЦикла;
//}_КОНСТРУКТОР_ДВИЖЕНИЙ_РЕГИСТРОВ

// Записываем движения регистров
Движения.Начисления.Записать();

// Получим список всех сотрудников, содержащихся в документе
Запрос = Новый Запрос(
"ВЫБРАТЬ РАЗЛИЧНЫЕ
| НачисленияСотрудникамНачисления.Сотрудник
| ИЗ
| Документ.НачисленияСотрудникам.Начисления КАК НачисленияСотрудникамНачисления
|
| ГДЕ
| НачисленияСотрудникамНачисления.Ссылка = &ТекущийДокумент");

Запрос.УстановитьПараметр("ТекущийДокумент", Ссылка);

// Сформируем список сотрудников
ТаблЗнач = Запрос.Выполнить().Выгрузить();
МассивСотрудников = ТаблЗнач.ВыгрузитьКолонку("Сотрудник");
```



```
// Вызов процедуры РассчитатьНачисления из общего модуля  
ПроведениеРасчетов.РассчитатьНачисления(Движения.Начисления,  
ПланыВидовРасчета.ОсновныеНачисления.Оклад, МассивСотрудников);  
Движения.Начисления.Записать( , Истина);
```

```
ПроведениеРасчетов.РассчитатьНачисления(Движения.Начисления, ПланыВидовРасчета.  
ОсновныеНачисления.Премия, МассивСотрудников);  
Движения.Начисления.Записать( , Истина);
```

КонецПроцедуры

Обратите внимание: при проведении документа мы сначала записываем движения, сформированные документом, в регистр (Движения.Начисления.Записать()), а затем передаем этот набор записей регистра в процедуру расчета РассчитатьНачисления(), которую мы создадим в общем модуле ПроведениеРасчетов.

Эту процедуру мы вызываем сначала для расчета первичных записей (Оклад), а затем для расчета вторичных (Премия).

Процедура расчета на основе описанных в ней алгоритмов и данных, содержащихся в записях регистра, должна сформировать значения ресурсов регистра.

После того как ресурсы будут рассчитаны, мы перезаписываем набор записей регистра без формирования записей перерасчета (второй параметр в методе Записать() — Истина).

Перед вызовом процедуры из общего модуля мы с помощью запроса формируем массив сотрудников, перечисленных в документе, чтобы передать его в вызываемую процедуру.

Для параметра запроса ТекущийДокумент устанавливаем значение стандартного реквизита документа — Ссылка. Используя метод запроса Запрос.Выполнить().Выгрузить(), выгружаем результат запроса в таблицу значений (переменную ТаблЗнач). Затем формируем массив МассивСотрудников, содержащий колонку Сотрудник из этой таблицы значений.

Теперь создадим в ветке Общие новый общий модуль ПроведениеРасчетов.

Установим флажок Вызов сервера для видимости его экспортных процедур и функций (рис. 18.13).




Рис. 18.13. Свойства общего модуля

Добавим в него заготовку процедуры РассчитатьНачисления (листинг 18.3).

Листинг 18.3. Заготовка процедуры «РассчитатьНачисления»

Процедура РассчитатьНачисления(НаборЗаписейРегистра, ТребуемыйВидРасчета,
СписокСотрудников) Экспорт

Регистратор = НаборЗаписейРегистра.Отбор.Регистратор.Значение;

// Рассчитать первичные записи

Если ТребуемыйВидРасчета = ПланыВидовРасчета.ОсновныеНачисления.Оклад Тогда

// Рассчитать вторичные записи

ИначеЕсли ТребуемыйВидРасчета = ПланыВидовРасчета.ОсновныеНачисления.Премия Тогда

КонецЕсли;

КонецПроцедуры

Алгоритм расчета начислений будет различным при расчете первичных (вид расчета – Оклад) и вторичных записей (вид расчета – Премия), и каждая из его частей будет находиться в своей ветке условия Если...

При расчете первичных записей нам понадобятся данные графика из регистра расчета, поэтому добавим в первую ветку условия запроса по виртуальной таблице регистра расчета РегистрРасчета.Начисления.ДанныеГрафика (листинг 18.4).

Листинг 18.4. Изменение процедуры «РассчитатьНачисления»

```

Процедура РассчитатьНачисления(НаборЗаписейРегистра, ТребуемыйВидРасчета,
                                СписокСотрудников) Экспорт

    Регистратор = НаборЗаписейРегистра.Отбор.Регистратор.Значение;

    // Рассчитать первичные записи
    Если ТребуемыйВидРасчета = ПланыВидовРасчета.ОсновныеНачисления.Оклад Тогда

        Запрос = Новый Запрос;
        Запрос.Текст =
            "ВЫБРАТЬ
             |   НачисленияДанныеГрафика.ЗначениеПериодДействия КАК Норма,
             |   НачисленияДанныеГрафика.ЗначениеФактическийПериодДействия КАК Факт,
             |   НачисленияДанныеГрафика.НомерСтроки КАК НомерСтроки
             |
             | ИЗ
             |   РегистрРасчета.Начисления.ДанныеГрафика(Регистратор = &Регистратор И
             |       ВидРасчета = &ВидРасчета И Сотрудник В (&СписокСотрудников))
             |   КАК НачисленияДанныеГрафика";

        Запрос.УстановитьПараметр("Регистратор", Регистратор);
        Запрос.УстановитьПараметр("ВидРасчета", ТребуемыйВидРасчета);
        Запрос.УстановитьПараметр("СписокСотрудников", СписокСотрудников);

        ВыборкаРезультата = Запрос.Выполнить().Выбрать();

    // Рассчитать вторичные записи
    ИначеЕсли ТребуемыйВидРасчета = ПланыВидовРасчета.ОсновныеНачисления.Премия Тогда

        КонецЕсли;

    КонецПроцедуры

```

В этом запросе мы выбираем из виртуальной таблицы данных графика регистра расчета значение графика для периода действия и для фактического периода действия. При задании параметров виртуальной таблицы мы ограничиваем выборку регистратором, нужным нам видом расчета и списком сотрудников, по которым нужно получить значения графика.

Теперь добавим обход переданного в процедуру набора записей и расчет записей, для которых получены значения графика (листинг 18.5).

Листинг 18.5. Добавление обхода набора записей и расчета первичных записей

```

Процедура РассчитатьНачисления(НаборЗаписейРегистра, ТребуемыйВидРасчета,
                                СписокСотрудников) Экспорт

    Регистратор = НаборЗаписейРегистра.Отбор.Регистратор.Значение;

    // Рассчитать первичные записи
    Если ТребуемыйВидРасчета = ПланыВидовРасчета.ОсновныеНачисления.Оклад Тогда
        ...
        ВыборкаРезультата = Запрос.Выполнить().Выбрать();

        Для Каждого ЗаписьРегистра Из НаборЗаписейРегистра Цикл
            СтруктураНомер = Новый Структура("НомерСтроки");
            СтруктураНомер.НомерСтроки = ЗаписьРегистра.НомерСтроки;
            ВыборкаРезультата.Сбросить();

            Если ВыборкаРезультата.НайтиСледующий(СтруктураНомер) Тогда

                Если ВыборкаРезультата.Норма = 0 Тогда
                    Сообщение = Новый СообщениеПользователю;
                    Сообщение.Текст = "Вид расчета: Оклад –
                                     Нет рабочих дней в заданном периоде";
                    Сообщение.Сообщить();
                    ЗаписьРегистра.Результат = 0;

                Иначе

                    // Рассчитать оклад по фактическому периоду и исходным данным
                    ЗаписьРегистра.Результат = (ЗаписьРегистра.ИсходныеДанные
                                                ВыборкаРезультата.Норма) * ВыборкаРезультата.Факт;
                    Сообщение = Новый СообщениеПользователю;
                    Сообщение.Текст = "Выполнен расчет " + ЗаписьРегистра.Регистратор
                                     + " – " + ЗаписьРегистра.ВидРасчета + " – " +
                                     ЗаписьРегистра.Сотрудник;
                    Сообщение.Сообщить();

                КонецЕсли;

            КонецЕсли;

        КонецЦикла;

    // Рассчитать вторичные записи
    ИначеЕсли ТребуемыйВидРасчета = ПланыВидовРасчета.ОсновныеНачисления.Премия Тогда

        КонецЕсли;

    КонецПроцедуры

```

Для каждой записи из набора записей регистра расчета мы получаем номер строки, идентифицирующий начисление для конкретного

сотрудника, и по этому номеру ищем соответствующую запись в выборке из результата запроса.

Если в результате запроса есть запись с таким номером строки, мы рассчитываем результат записи регистра расчета. То есть мы получаем начисление по окладу для каждого сотрудника как результат от деления начисленной суммы (поле регистра ИсходныеДанные) на количество рабочих дней в месяце (Норма) и умножения на фактически отработанные рабочие дни (Факт).

Добавим текст запроса во вторую ветку условия Если... с той лишь разницей, что теперь мы будем получать значения базы, используя виртуальную таблицу регистра расчета РегистрРасчета.Начисления.БазаНачисления (листинг 18.6).

Листинг 18.6. Добавление текста запроса во вторую ветку условия

```
Процедура РассчитатьНачисления(НаборЗаписейРегистра, ТребуемыйВидРасчета,
                                СписокСотрудников) Экспорт

    Регистратор = НаборЗаписейРегистра.Отбор.Регистратор.Значение;

    // Рассчитать первичные записи
    Если ТребуемыйВидРасчета = ПланыВидовРасчета.ОсновныеНачисления.Оклад Тогда
...
    // Рассчитать вторичные записи
    ИначеЕсли ТребуемыйВидРасчета = ПланыВидовРасчета.ОсновныеНачисления.Премия Тогда

        Запрос = Новый Запрос;
        Запрос.Текст =
            "ВЫБРАТЬ
              НачисленияБазаНачисления.РезультатБаза КАК База,
              НачисленияБазаНачисления.НомерСтроки КАК НомерСтроки
            ИЗ
              РегистрРасчета.Начисления.БазаНачисления(&ИзмеренияОсновного,
                &ИзмеренияБазового, , Регистратор =
                  &Регистратор И ВидРасчета = &ВидРасчета И
                  Сотрудник В (&СписокСотрудников))
              КАК НачисленияБазаНачисления";

        Измер = Новый Массив(1);
        Измер[0] = "Сотрудник";

        Запрос.УстановитьПараметр("ИзмеренияОсновного", Измер);
        Запрос.УстановитьПараметр("ИзмеренияБазового", Измер);
        Запрос.УстановитьПараметр("Регистратор", Регистратор);
        Запрос.УстановитьПараметр("ВидРасчета", ТребуемыйВидРасчета);
        Запрос.УстановитьПараметр("СписокСотрудников", СписокСотрудников);

        ВыборкаРезультата = Запрос.Выполнить().Выбрать();

    КонецЕсли;

КонецПроцедуры
```

В параметрах виртуальной таблицы запроса мы кроме привычных для нас регистратора, вида расчета и списка сотрудников задаем еще измерения основного и базового регистров. В нашем случае это будет один и тот же регистр Начисления, а нужное нам измерение – Сотрудник.

В заключение осталось добавить во второе условие Если... обход набора записей регистра расчета и вычисление результата вторичных записей (листинг 18.7).

Листинг 18.7. Добавление обхода набора записей регистра и вычисления результата вторичных записей

```
Процедура РассчитатьНачисления(НаборЗаписейРегистра, ТребуемыйВидРасчета,
                                СписокСотрудников) Экспорт

    Регистратор = НаборЗаписейРегистра.Отбор.Регистратор.Значение;

    // Рассчитать первичные записи
    Если ТребуемыйВидРасчета = ПланыВидовРасчета.ОсновныеНачисления.Оклад Тогда
...
    // Рассчитать вторичные записи
    ИначеЕсли ТребуемыйВидРасчета = ПланыВидовРасчета.ОсновныеНачисления.Премия Тогда

        ВыборкаРезультата = Запрос.Выполнить().Выбрать();

        Для Каждого ЗаписьРегистра Из НаборЗаписейРегистра Цикл
            СтруктураНомер = Новый Структура("НомерСтроки");
            СтруктураНомер.НомерСтроки = ЗаписьРегистра.НомерСтроки;
            ВыборкаРезультата.Сбросить();

            Если ВыборкаРезультата.НайтиСледующий(СтруктураНомер) Тогда
                ЗаписьРегистра.Результат = ВыборкаРезультата.База * (10 / 100);
                Сообщение = Новый СообщениеПользователю;
                Сообщение.Текст = "Выполнен расчет " + ЗаписьРегистра.Регистратор + " – " +
                    ЗаписьРегистра.ВидРасчета + " – " + ЗаписьРегистра.Сотрудник;
                Сообщение.Сообщить();

            КонецЕсли;

        КонецЦикла;

    КонецЕсли;

КонецПроцедуры
```

Сумму начисленной премии мы рассчитываем как 10 % от рассчитанной оплаты по окладу.

В режиме «1С:Предприятие»

Запустим «1С:Предприятие» в режиме отладки и проверим правильность работы процедуры расчета.

Отменим проведение документа Начисления сотрудникам № 3 (Еще ► Отмена проведения) и перепроведем документы Начисления сотрудникам № 1 и № 2. Регистр расчета Начисления должен выглядеть следующим образом (рис. 18.14, 18.15).

Период регистрации	Регистратор	Н.	Вид расчета	Сторно	Сотрудник	Результат	График работы
01.07.2013 0:00:00	Начисление с...	1	Оклад		Гусаков Николай Дмитриевич	7 000,00	График мастеров
01.07.2013 0:00:00	Начисление с...	2	Оклад		Деловой Иван Сергеевич	8 000,00	График мастеров
01.07.2013 0:00:00	Начисление с...	3	Оклад		Симонов Валерий Михайлович	3 000,00	График мастеров
01.07.2013 0:00:00	Начисление с...	1	Премия		Гусаков Николай Дмитриевич	700,00	График мастеров
01.07.2013 0:00:00	Начисление с...	2	Премия		Деловой Иван Сергеевич	800,00	График мастеров

Рис. 18.14. Записи регистра «Начисления»

Исходные данные	Период действия	Дата начала периода дей...	Дата окончания периода дей...	Дата начала базового пери...	Дата окончания базового пери...
7 000,00	01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59
8 000,00	01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59
3 000,00	01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59
	01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59
	01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59

Рис. 18.15. Записи регистра «Начисления»

Мы видим, что всем сотрудникам произведены начисления по окладу (поле Результат) за полный месяц в соответствии с исходными данными (поле Исходные данные).

Сотрудникам Гусакову и Деловому начислена премия в размере 10 % от суммы начисления по окладу.

Проведем документ Начисление сотрудникам № 3, а затем № 1 и № 2.

При этом отчет Перерасчет должен быть пуст.

Состояние регистра изменится следующим образом (рис. 18.16, 18.17).

Период регистрации	Регистратор	Н...	Вид расчета	Сторно	Сотрудник	Результат	График работы
01.07.2013 0:00:00	Начисление с...	1	Оклад		Гусаков Николай Дмитриевич	4 565,22	График мастеров
01.07.2013 0:00:00	Начисление с...	2	Оклад		Деловой Иван Сергеевич	8 000,00	График мастеров
01.07.2013 0:00:00	Начисление с...	3	Оклад		Симонов Валерий Михайлович	3 000,00	График мастеров
01.07.2013 0:00:00	Начисление с...	1	Премия		Гусаков Николай Дмитриевич	456,52	График мастеров
01.07.2013 0:00:00	Начисление с...	2	Премия		Деловой Иван Сергеевич	800,00	График мастеров
01.07.2013 0:00:00	Начисление с...	1	Невыход		Гусаков Николай Дмитриевич		График мастеров

Рис. 18.16. Записи регистра «Начисления»

Исходные данные	Период действия	Дата начала периода дей...	Дата окончания периода дей...	Дата начала базового пери...	Дата окончания базового перис...
7 000,00	01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59
8 000,00	01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59
3 000,00	01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59
	01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59
	01.07.2013 0:00:00	01.07.2013 0:00:00	31.07.2013 23:59:59	01.07.2013 0:00:00	31.07.2013 23:59:59
	01.07.2013 0:00:00	01.07.2013 0:00:00	10.07.2013 23:59:59	01.07.2013 0:00:00	10.07.2013 23:59:59

Рис. 18.17. Записи регистра «Начисления»

В результате невыхода Гусакова на работу сумма оплаты по окладу будет уменьшена, и соответствующим образом уменьшится начисленная ему премия.

Отчет о начислениях сотрудникам

Теперь мы посмотрим, каким образом можно использовать данные, хранящиеся в регистре расчета, для получения в отчете итоговой информации о начислениях сотрудникам (рис. 18.18).

Начисления сотрудникам					Результат
Сотрудник	Вид расчета	Начало	Окончание	Регистратор	
Гусаков Николай Дмитриевич	Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:28:40	5 021,74
	Премия	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 2 от 24.07.2013 20:07:48	456,52
	Невыход	01.07.2013 0:00:00	10.07.2013 23:59:59	Начисление сотрудникам 3 от 24.07.2013 20:16:44	
					8 800,00
Деловой Иван Сергеевич	Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:28:40	8 000,00
	Премия	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 2 от 24.07.2013 20:07:48	800,00
Симонов Валерий Михайлович	Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:28:40	3 000,00
					3 000,00
Итого					16 821,74

Рис. 18.18. Результат отчета

В режиме «Конфигуратор»

Создадим в конфигураторе новый объект конфигурации Отчет. Назовем его НачисленияСотрудникам. Создадим основную схему компоновки данных отчета, добавим новый Набор данных – запрос, откроем конструктор запроса.

Запрос для набора данных

Выберем таблицу регистра расчета Начисления (рис. 18.19).




Рис. 18.19. Состав полей таблицы «Начисления»

Из нее выберем следующие поля:

- Сотрудник,
- ВидРасчета,
- ПериодДействияНачало,
- ПериодДействияКонец,
- Регистратор,
- Результат (рис. 18.20).




Рис. 18.20. Выбранные поля

На закладке Объединения/Псевдонимы определим следующие псевдонимы полей ПериодДействияНачало и ПериодДействияКонец (рис. 18.21).

Имя поля	Запрос 1
↳ Сотрудник	↳ Начисления.Сотрудник
↳ ВидРасчета	↳ Начисления.ВидРасчета
↳ Начало	↳ Начисления.Период.Действия.Начало
↳ Окончание	↳ Начисления.Период.Действия.Конец
↳ Регистратор	↳ Начисления.Регистратор
↳ Результат	↳ Начисления.Результат

Рис. 18.21. Объединения/Псевдонимы

На этом создание запроса закончено, нажмем ОК.

Ресурсы

Перейдем на закладку Ресурсы и укажем, что должна быть рассчитана сумма по полю Результат.

Настройки

После этого перейдем на закладку Настройки и создадим структуру отчета.

Добавим группировку по полю Сотрудник и в ней – подчиненную группировку детальных записей.

В качестве полей, выводимых в отчет, выберем поля ВидРасчета, Начало, Окончание, Регистратор и Результат.

В результате окно настроек отчета должно иметь вид (рис. 18.22).




Рис. 18.22. Структура и выбранные поля отчета

На закладке Сортировка укажем, что сортировка должна выполняться по возрастанию значения поля Сотрудник и Регистратор (рис. 18.23).




Рис. 18.23. Настройка сортировки отчета

И в заключение на закладке Другие настройки зададим заголовок отчета – Начисления сотрудникам.

На этом создание схемы компоновки данных закончено, закроем ее.

В окне редактирования объекта конфигурации Отчет НачисленияСотрудникам на закладке подсистемы укажем, что отчет будет вызываться из подсистем РасчетЗарплаты и Бухгалтерия.

В режиме «1С:Предприятие»

Запустим «1С:Предприятие» в режиме отладки.

В командной панели раздела Расчет зарплаты выполним команду Начисления сотрудникам и сформируем отчет (рис. 18.24).

← → ☆ Начисления сотрудникам

Сформировать Выбрать вариант... Еще ▾

Начисления сотрудникам

Сотрудник	Вид расчета	Начало	Окончание	Регистратор	Результат
Гусakov Николай Дмитриевич	Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:28:40	5 021,74
	Премия	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 2 от 24.07.2013 20:07:48	4 565,22
	Начисл.	01.07.2013 0:00:00	10.07.2013 23:59:59	Начисление сотрудникам 3 от 24.07.2013 20:16:44	456,62
Деловой Иван Сергеевич	Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:28:40	8 800,00
	Премия	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 2 от 24.07.2013 20:07:48	8 000,00
Омников Валерий Михайлович	Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:28:40	3 000,00
Итого					16 821,74

Рис. 18.24. Результат отчета

Перерасчет

Итак, в нашем алгоритме работы с данными расчета осталось одно узкое место – контроль актуальности данных, содержащихся в регистре расчета.

До сих пор мы с вами использовали служебный отчет Перерасчет для того, чтобы определить, являются ли данные в регистре расчета актуальными (если отчет пуст), или же они требуют перерасчета.

Теперь мы с вами создадим специальную процедуру, которая будет определять, требуется ли перерасчет данных регистра расчета, и, если такая необходимость есть, выполнять перерасчет.

Поскольку единственным способом получения итоговой информации о начислениях сотрудникам в нашей конфигурации является отчет НачисленияСотрудникам, для вызова этой процедуры мы создадим форму этого отчета и добавим в командную панель формы кнопку Перерассчитать, по которой будет выполняться перерасчет данных регистра.

В режиме «Конфигуратор»

Для этого в окне редактирования объекта конфигурации Отчет НачисленияСотрудникам перейдем на закладку Формы, нажмем кнопку открытия  и создадим основную форму отчета.

Затем в правом верхнем окне редактора форм перейдем на закладку Команды и на закладке Команды формы создадим команду формы Перерассчитать (рис. 18.25).

Теперь нужно установить Действие для этой команды.

Для этого нажмем кнопку открытия  в строке Действие.

На запрос конфигулятора о типе обработчика команды ответим, что мы хотим создать обработчик, выполняющийся на клиенте.

В модуле формы будет создан шаблон клиентской процедуры Перерассчитать(), в которую мы поместим вызов процедуры ПерерассчитатьНачисления() из общего модуля ПроведениеРасчетов (листинг 18.8).

Саму процедуру перерасчета поместим в общем модуле ПроведениеРасчетов (листинг 18.9).




Рис. 18.25. Добавление команды формы

Листинг 18.8. Текст обработчика команды «Перерассчитать»

```

&НаКлиенте
Процедура Перерассчитать(Команда)
    ПроведениеРасчетов.ПерерассчитатьНачисления(ПредопределенноеЗначение(
        "ПланВидовРасчета.ОсновныеНачисления.Оклад"));
    ПроведениеРасчетов.ПерерассчитатьНачисления(ПредопределенноеЗначение(
        "ПланВидовРасчета.ОсновныеНачисления.Премия"));
КонецПроцедуры

```

Листинг 18.9. Процедура перерасчета начислений

Процедура ПерерассчитатьНачисления(ТребуемыйВидРасчета) Экспорт

```

// Здесь следует выбрать из набора записей перерасчета записи в следующей последовательности:
// записи документа1 для сотрудников из списка,
// записи документа2 для сотрудников из списка и т. д.
Запрос = Новый Запрос(
    "ВЫБРАТЬ
        НачисленияПерерасчет.ОбъектПерерасчета,
        НачисленияПерерасчет.Сотрудник
    ИЗ
        РегистрРасчета.Начисления.Перерасчет КАК НачисленияПерерасчет
    ГДЕ
        НачисленияПерерасчет.ВидРасчета = &ТребуемыйВидРасчета
    ИТОГИ ПО
        НачисленияПерерасчет.ОбъектПерерасчета");

```

```

Запрос.УстановитьПараметр("ТребуемыйВидРасчета", ТребуемыйВидРасчета);
СписокСотрудников = Новый СписокЗначений;

// Перебрать группировку по регистратору.
ВыборкаПоРегистратору = Запрос.Выполнить().Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам);
Пока ВыборкаПоРегистратору.Следующий() Цикл
    Регистратор = ВыборкаПоРегистратору.ОбъектПерерасчета;

    // Перебрать группировку по сотрудникам для выбранного регистратора
    // и создать список сотрудников.
    ВыборкаПоСотрудникам = ВыборкаПоРегистратору.Выбрать();
    СписокСотрудников.Очистить();
    Пока ВыборкаПоСотрудникам.Следующий() Цикл
        СписокСотрудников.Добавить(ВыборкаПоСотрудникам.Сотрудник);
    КонецЦикла;

    // Получить набор записей регистра расчета для выбранного регистратора.
    НаборЗаписей = РегистрыРасчета.Начисления.СоздатьНаборЗаписей();
    НаборЗаписей.Отбор.Регистратор.Значение = Регистратор;
    НаборЗаписей.Прочитать();

    РассчитатьНачисления(НаборЗаписей, ТребуемыйВидРасчета, СписокСотрудников);
    НаборЗаписей.Записать( , Истина);

    // Очистить перерассчитанные записи в перерасчете.
    НаборЗаписейПерерасчета =
        РегистрыРасчета.Начисления.Перерасчеты.Перерасчет.СоздатьНаборЗаписей();
    НаборЗаписейПерерасчета.Отбор.ОбъектПерерасчета.Значение = Регистратор;
    НаборЗаписейПерерасчета.Записать();
КонецЦикла;

КонецПроцедуры

```

В самом начале процедуры мы запросом выбираем данные о записях перерасчетов, содержащие переданный вид расчета и сгруппированные по объекту перерасчета.

Далее при обходе результата запроса мы формируем для каждого объекта перерасчета список сотрудников, читаем соответствующие записи регистра расчета и вызываем процедуру `РассчитатьНачисления()`, которая использовалась нами при расчете записей документа `НачисленияСотрудникам`.

После того как расчет записей выполнен, мы записываем набор записей без формирования записей перерасчета и очищаем записи перерасчета по тому объекту перерасчета, который только что обработали.

Вернемся в форму отчета `НачислениеСотрудникам`.

Итак, мы указали для команды Перерассчитать действие, то есть процедуру для ее выполнения. Но чтобы можно было воспользоваться этой командой, нужно создать в форме кнопку и связать ее с этой командой (в строке ИмяКоманды).

Проще всего это сделать перетаскиванием команды из окна Команды формы в окно элементов формы.

Перетащим мышью команду Перерассчитать в группу элементов формы ОсновнаяКоманднаяПанель.

При этом в форме появится кнопка Перерассчитать, а связь кнопки с командой будет установлена автоматически (рис. 18.26).




Рис. 18.26. Добавление кнопки в форму

В режиме «1С:Предприятие»

Запустим «1С:Предприятие» и проверим, как выполняется перерасчет записей регистра расчета.

Отменим проведение всех документов Начисления сотрудникам и проведем документ Начисления сотрудникам № 1 и затем № 2.

Сформируем отчет Начисления сотрудникам (рис. 18.27).

Теперь откроем документ Начисления сотрудникам № 1, изменим оклад Гусакова на 10 000 и проведем документ.

В отчете Начисления сотрудникам нажмем кнопку Перерассчитать.

Сформировать

Выбрать вариант...

Перерассчитать

Еще ▾

Начисления сотрудникам

Сотрудник					Результат
Вид расчета	Начало	Окончание	Регистратор		
Гусаков Николай Дмитриевич					7 700,00
Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:26:40		7 000,00
Премия	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 2 от 24.07.2013 20:07:48		700,00
Деловой Иван Сергеевич					8 800,00
Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:26:40		8 000,00
Премия	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 2 от 24.07.2013 20:07:48		800,00
Симонов Валерий Михайлович					3 000,00
Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:26:40		3 000,00
Итого					19 500,00

Рис. 18.27. Отчет «Начисления сотрудникам»

Будет выполнен перерасчет начисления премии Гусакову и Деловому (рис. 18.28).

Сообщения:		×
— Выполнен расчет Начисление сотрудникам 2 от 24.07.2013 20:07:48 – Премия – Гусаков Николай Дмитриевич		
— Выполнен расчет Начисление сотрудникам 2 от 24.07.2013 20:07:48 – Премия – Деловой Иван Сергеевич		

Рис. 18.28. Окно служебных сообщений

Чтобы увидеть в отчете актуальные данные, нажмем кнопку Сформировать.

Результат работы отчета будет содержать новые значения премии Гусакова (рис. 18.29).

Сформировать

Выбрать вариант...

Перерассчитать

Еще ▾

Начисления сотрудникам

Сотрудник				Результат
Вид расчета	Начало	Окончание	Регистратор	
Гусаков Николай Дмитриевич				11 000,00
Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:26:40	10 000,00
Премия	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 2 от 24.07.2013 20:07:48	1 000,00
Деловой Иван Сергеевич				8 800,00
Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:26:40	8 000,00
Премия	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 2 от 24.07.2013 20:07:48	800,00
Симонев Валерий Михайлович				3 000,00
Оклад	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:26:40	3 000,00
Итого				22 800,00

Рис. 18.29. Отчет «Начисления сотрудникам»

И, наконец, проведем документ Начисления сотрудникам № 3 и нажмем Перерассчитать в отчете Начисления сотрудникам.

Снова будет произведен перерасчет оклада и премии Гусакова (рис. 18.30).




Рис. 18.30. Окно служебных сообщений

Затем нажмем кнопку Сформировать. Данные отчета будут содержать актуальные значения начисления оклада и премии (рис. 18.31).

← → ☆ Начисления сотрудникам

Сформировать Выбрать вариант... Перерассчитать Еще ▾

Вид расчета	Сотрудник	Начало	Окончание	Регистратор	Результат
Оклад	Гусakov Николай Дмитриевич	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:28:40	7 173,91
Премия	Гусakov Николай Дмитриевич	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 2 от 24.07.2013 20:07:48	6 521,74
Невыгод	Гусakov Николай Дмитриевич	01.07.2013 0:00:00	10.07.2013 23:59:59	Начисление сотрудникам 3 от 24.07.2013 20:16:44	652,17
Оклад	Деловой Иван Сергеевич	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:28:40	8 800,00
Премия	Деловой Иван Сергеевич	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 2 от 24.07.2013 20:07:48	8 000,00
Оклад	Симонов Валерий Михайлович	01.07.2013 0:00:00	31.07.2013 23:59:59	Начисление сотрудникам 1 от 24.07.2013 19:28:40	3 000,00
Итого					18 973,91

Рис. 18.31. Отчет «Начисления сотрудникам»

Диаграмма Ганта

В конце этой главы мы совместим приятное с полезным и создадим с вами отчет, который в графическом виде будет показывать нам фактический период действия записей расчета.

Помимо наглядной демонстрации работы механизма вытеснения записей по периоду действия этот отчет позволит нам познакомиться с элементом формы, позволяющим создавать *диаграммы Ганта*.

Диаграмма Ганта представляет собой диаграмму интервалов на шкале времени (рис. 18.32) и отражает использование объектами (точками) ресурсов (серий).

Чтобы проще было представить себе составные части диаграммы Ганта, изучим диаграмму, которая должна получиться в результате работы создаваемого нами отчета.

Как мы уже говорили, эта диаграмма будет отображать для каждого сотрудника фактический период действия записи по каждому из видов расчета, имеющих место для этого сотрудника.




Рис. 18.32. Пример диаграммы Ганта

Итак, диаграмма Ганта представляет собой совокупность точек, серий и значений для каждой пары точка-серия.

В нашем случае точками диаграммы являются сотрудники, а сериями – виды расчетов. Таким образом, для каждого сотрудника существует некоторое значение диаграммы по каждой из серий, то есть по каждому из видов расчета.

Значение диаграммы Ганта представляет собой специальный объект, который автоматически формируется системой на основании того, какие точки и какие серии определены для данной диаграммы.

Этот объект является совокупностью (коллекцией) интервалов, то есть может содержать не один, а несколько интервалов, которые соответствуют паре серия-точка (создаваемый по умолчанию объект ЗначениеДиаграммыГанта не содержит ни одного интервала). Разработчик может получить значение диаграммы, указав интересующую его точку и серию, и затем добавить в коллекцию необходимое количество интервалов.

Все интервалы всех значений диаграммы располагаются с привязкой к единой оси времени, что дает возможность видеть их взаимное расположение.

Теперь коротко объясним последовательность наших дальнейших действий.

В качестве исходных данных для построения такой диаграммы мы возьмем данные регистра расчета Начисления. Каждая запись этого регистра уже содержит все необходимое для построения диаграммы: сотрудника, вид расчета, начало и конец интервала.

Нам останется только средствами встроенного языка разместить все это в диаграмме.

Итак, приступим.

В режиме «Конфигуратор»

Создадим новый объект конфигурации Отчет и назовем его ДиаграммаНачислений.

Для этого отчета мы не будем создавать схему компоновки данных, а создадим основную форму отчета и обеспечим формирование и настройку диаграммы Ганта с помощью кода на встроенном языке. В окне редактирования объекта конфигурации Отчет ДиаграммаНачислений перейдем на закладку Формы, нажмем кнопку открытия  и создадим основную форму отчета.

В правом верхнем окне редактора форм на закладке Реквизиты находятся реквизиты формы. Мы видим здесь основной реквизит формы Отчет, который был создан автоматически при создании формы.

Нажмем кнопку Добавить и добавим новый реквизит формы. Назовем его ДиаграммаГанта и выберем его тип ДиаграммаГанта (рис. 18.33).

Теперь перетащим новый реквизит в окно элементов формы, которое пока пусто.

В окне элементов формы будет создано новое поле для отображения диаграммы Ганта, а в нижнем окне просмотра формы мы сразу увидим поле диаграммы (рис. 18.34).




Рис. 18.33. Добавление реквизита формы




Рис. 18.34. Добавление диаграммы Ганта в форму

На закладке Команды создадим команду формы Сформировать (рис. 18.35).




Рис. 18.35. Добавление команды формы

Теперь нужно установить Действие для этой команды.

Для этого нажмем кнопку открытия  в строке Действие.

На запрос конфигулятора о типе обработчика команды ответим, что мы хотим создать клиентский обработчик команды формы с вызовом из него процедуры, выполняющейся на сервере без контекста формы (рис. 18.36).




Рис. 18.36. Выбор типа обработчика команды формы

В модуле формы будут созданы шаблоны двух процедур: клиентской процедуры `Сформировать()` и серверной внеконтекстной процедуры `СформироватьНаСервере()`, которая вызывается из процедуры `Сформировать()`.

Мы не будем сейчас подробно рассматривать, что такое серверная внеконтекстная процедура. Отметим только, что внеконтекстная процедура выполняется на сервере значительно быстрее за счет того, что на сервер с клиента не передается весь контекст формы.

Однако нам нужно передать в процедуру `СформироватьНаСервере()` в качестве параметра ссылку на реквизит формы `ДиаграммаГанта`, чтобы на сервере заполнить его данными. Поэтому изменим текст модуля следующим образом (листинг 18.10).

Листинг 18.10. Текст обработчика команды «Сформировать»

```
&НаКлиенте
Процедура Сформировать(Команда)
    СформироватьНаСервере(ДиаграммаГанта);
КонецПроцедуры

&НаСервереБезКонтекста
Процедура СформироватьНаСервере(Диаграмма)
    // Вставить содержимое обработчика.
КонецПроцедуры
```

В процедуру `СформироватьНаСервере()` мы вставим заготовку запроса (листинг 18.11).

Листинг 18.11. Процедура «СформироватьНаСервере()»

```
&НаСервереБезКонтекста
Процедура СформироватьНаСервере(Диаграмма)

    Запрос = Новый Запрос;
    Запрос.Текст = ;

КонецПроцедуры
```

Установим курсор перед точкой с запятой, вызовем контекстное меню, откроем конструктор запроса и создадим новый запрос.

Выберем виртуальную таблицу регистра расчета `Начисления.ФактическийПериодДействия`.

Из этой таблицы выберем следующие поля (рис. 18.37):

- `Сотрудник`,
- `ВидРасчета`,
- `ПериодДействияНачало`,
- `ПериодДействияКонец`,

- Результат,
- Регистратор,
- Регистратор.Представление.




Рис. 18.37. Выбранные поля

Все, запрос готов.

Теперь нажмем ОК и после текста запроса добавим в процедуру следующий текст (листинг 18.12).

Листинг 18.12. Процедура «СформироватьНаСервере()»

&НаСервереБезКонтекста

Процедура СформироватьНаСервере(Диаграмма)

Запрос = Новый Запрос;

Запрос.Текст =

"ВЫБРАТЬ

НачисленияФактическийПериодДействия.Сотрудник,
НачисленияФактическийПериодДействия.ВидРасчета,
НачисленияФактическийПериодДействия.ПериодДействияНачало,
НачисленияФактическийПериодДействия.ПериодДействияКонец,
НачисленияФактическийПериодДействия.Результат,
НачисленияФактическийПериодДействия.Регистратор,
НачисленияФактическийПериодДействия.Регистратор.Представление

ИЗ

РегистрРасчета.Начисления.ФактическийПериодДействия
КАК НачисленияФактическийПериодДействия";

ВыборкаРезультата = Запрос.Выполнить().Выбрать();

// Запретить обновление диаграммы.

Диаграмма.Обновление = Ложь;

Диаграмма.Очистить();

Диаграмма.ОтображатьЗаголовок = Ложь;

// Заполнить диаграмму.

Пока ВыборкаРезультата.Следующий() Цикл

// Получить серию, точку и значение для них.

ТекущаяСерия = Диаграмма.УстановитьСерию(ВыборкаРезультата.ВидРасчета);

```
ТекущаяТочка = Диаграмма.УстановитьТочку(ВыборкаРезультата.Сотрудник);
ТекущееЗначение = Диаграмма.ПолучитьЗначение(ТекущаяТочка, ТекущаяСерия);
```

```
// Создать нужные нам интервалы в значении.
ТекущийИнтервал = ТекущееЗначение.Добавить();
ТекущийИнтервал.Начало = ВыборкаРезультата.ПериодДействияНачало;
ТекущийИнтервал.Конец = ВыборкаРезультата.ПериодДействияКонец;
ТекущийИнтервал.Текст = ВыборкаРезультата.РегистраторПредставление;
ТекущийИнтервал.Расшифровка = ВыборкаРезультата.Регистратор;
```

```
КонецЦикла;
```

```
// Раскрасить серии своими цветами.
```

```
Для Каждого Серии из Диаграмма.Серии Цикл
```

```
Если Серия.Значение = ПланыВидовРасчета.ОсновныеНачисления.Оклад Тогда
    Серия.Цвет = WEBЦвета.Желтый;
```

```
ИначеЕсли Серия.Значение = ПланыВидовРасчета.ОсновныеНачисления.Премия Тогда
    Серия.Цвет = WEBЦвета.Зеленый;
```

```
ИначеЕсли Серия.Значение = ПланыВидовРасчета.ОсновныеНачисления.Невыход Тогда
    Серия.Цвет = WEBЦвета.Красный;
```

```
КонецЕсли;
```

```
КонецЦикла;
```

```
// Разрешить обновление диаграммы.
```

```
Диаграмма.Обновление = Истина;
```

```
КонецПроцедуры
```

Сначала мы запрещаем обновление диаграммы на то время, пока мы будем заполнять ее данными. Это нужно для того, чтобы в процессе заполнения не выполнялись пересчеты при каждом изменении данных диаграммы. После окончания заполнения диаграммы мы разрешим обновление, и все пересчеты будут выполнены один раз.

Затем в цикле по выборке запроса мы заполняем диаграмму.

Сначала, используя методы `УстановитьСерию()` и `Установить Точку()`, мы получаем либо существующие, либо новые точку и серию. Точки и серии однозначно идентифицируются своими значениями, в качестве которых мы используем сотрудника и вид расчета из результата запроса.

После того как точка и серия получены, с помощью метода `ПолучитьЗначение()` мы получаем соответствующее им значение диаграммы.

Затем мы добавляем в значение диаграммы новый интервал, задаем его начало и конец, задаем текст интервала, который будет показываться во всплывающей подсказке, и задаем расшифровку интервала, которая будет выполняться при двойном щелчке мышью на этом интервале.

После того как все значения диаграммы сформированы, мы раскрашиваем серии своими цветами. Серии диаграммы представляют собой коллекцию значений, которую мы перебираем при помощи конструкции Для Каждого ... Цикл.

Теперь вернемся в форму и добавим в нее кнопку для выполнения команды Сформировать.

Для этого перетащим мышью команду Сформировать из окна Команды формы в окно элементов формы (рис. 18.38).




Рис. 18.38. Добавление кнопки в форму

В заключение в окне редактирования объекта конфигурации Отчет ДиаграммаНачислений на закладке подсистемы укажем, что отчет будет вызываться из подсистемы РасчетЗарплаты.

В режиме «1С:Предприятие»

Запустим «1С:Предприятие» в режиме отладки и посмотрим на результат работы отчета (рис. 18.39).




Рис. 18.39. Отчет «Диаграмма начислений»

А теперь посмотрим, как выглядит механизм вытеснения по периоду действия на конкретном примере.

Откроем документ Начисления сотрудникам № 3 и вместо одного прогула с 1 по 10 число зададим Гусakovу два прогула: с 3 по 7 число и с 12 по 15 число.

Проведем документ и снова нажмем Сформировать в нашем отчете (рис. 18.40).




Рис. 18.40. Отчет «Диаграмма начислений»

Теперь вы наглядно видите, как записи вида расчета Невыход вытеснили по периоду действия запись расчета Оклад, изменив ее фактический период действия.

Следует отметить, что существует также возможность интерактивной настройки параметров диаграммы Ганта в режиме 1С:Предприятие, доступная через пункт контекстного меню Настройка... Также настройку параметров диаграммы, таких как Отображать заголовок, Отображать легенду, Прозрачный фон и т.п., можно выполнить в конфигураторе, в редакторе формы отчета Диаграмма начислений. Для этого в панели свойств реквизита, содержащего диаграмму, нужно выполнить команду Настройка Открыть.

Контрольные вопросы

- ☒ Как создать движения документа по регистру расчета?
- ☒ Как запросом получить записи перерасчета?
- ☒ Как работает перерасчет?
- ☒ Как рассчитать записи регистра расчета?
- ☒ Как запросом получить данные графика и базы?
- ☒ Как выполнить перерасчет отдельных записей регистра расчета?
- ☒ Как получить запросом записи регистра расчета?
- ☒ Как получить запросом фактический период действия записей регистра расчета?
- ☒ Для чего используется диаграмма Ганта?
- ☒ Как устроена диаграмма Ганта?
- ☒ Как заполнить диаграмму Ганта данными?